Some MAPLE exercises for M302 - Differential Equations

By Peter Monk (monk@math.udel.edu), May 1995 Version of February 16, 1996

Contents

1 Introduction 4 1.1 General Caveat . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4 1.2 Sources and Acknowledgments . . . . . . . . . . . . . . . . . . . . . . . . . . 4

2 Unix Account Setup 5 2.1 Activating your UD computer account . . . . . . . . . . . . . . . . . . . . . 5 2.2 Setting up your UD account . . . . . . . . . . . . . . . . . . . . . . . . . . . 5 2.3 Working on MATH302-013class assignments outside the Math. Lab. . . . . . 6 2.4 Using Maple from X-terminals . . . . . . . . . . . . . . . . . . . . . . . . . . 7 2.4.1 Activating your computer account . . . . . . . . . . . . . . . . . . . . 7 2.4.2 Setting up the M302 environment . . . . . . . . . . . . . . . . . . . . 7 2.4.3 Starting Maple . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 7 2.4.4 Logging out from an X-terminal . . . . . . . . . . . . . . . . . . . . . 8

3 Integral Curves and Direction Fields 9 3.1 Introduction . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 9 3.2 Getting started in the math lab . . . . . . . . . . . . . . . . . . . . . . . . . 9 3.3 Drawing integral curves. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 10 3.4 Drawing direction fields. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 11

4 Solving ODE’s in MAPLE 13 4.1 Solving ODEs in MAPLE exactly . . . . . . . . . . . . . . . . . . . . . . . . 13 4.2 Solving ODEs implicitly and drawing integral curves . . . . . . . . . . . . . 15 4.3 Approximate solutions via dsolve . . . . . . . . . . . . . . . . . . . . . . . . 16

5 First Assignment (for credit) 17 5.1 Population models. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 17 5.2 The Falling Raindrop . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 18 6 Homogeneous second order ordinary differential equations 20 6.1 Linearity and independence of solutions of homogeneous problems . . . . . . 20 6.1.1 Superposition principle . . . . . . . . . . . . . . . . . . . . . . . . . . 20

6.1.2 Linear independence . . . . . . . . . . . . . . . . . . . . . . . . . . . 21 6.2 Solving Second Order Equations . . . . . . . . . . . . . . . . . . . . . . . . . 23 6.2.1 Using the characteristic equation . . . . . . . . . . . . . . . . . . . . 23 6.2.2 Using dsolve . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 23 6.3 Complex roots of the Characteristic equation . . . . . . . . . . . . . . . . . . 25

7 Second Assignment (for credit) 27 7.1 Second order linear ordinary differential equations . . . . . . . . . . . . . . . 27

8 Inhomogeneous Linear Problems 29 8.1 The Method of Undetermined Coefficients . . . . . . . . . . . . . . . . . . . 29 8.2 Variation of Parameters . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 31 8.3 Homogeneous Second Order Problems . . . . . . . . . . . . . . . . . . . . . . 33 8.3.1 No damping . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 33 8.3.2 Negative damping . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 34 8.3.3 Positive Damping . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 34 8.3.4 Phase Angles . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 35 8.4 Assignment on the solution of inhomogeneous second order problems . . . . 37 8.4.1 Inhomogeneous problems . . . . . . . . . . . . . . . . . . . . . . . . . 37 8.4.2 Resonance . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 37 9 Series Solutions 38 9.1 Series solutions in dsolve . . . . . . . . . . . . . . . . . . . . . . . . . . . . 38 9.2 Taking more and more terms . . . . . . . . . . . . . . . . . . . . . . . . . . . 39

10 The Laplace Transform 41 10.1 Computing the Laplace transform . . . . . . . . . . . . . . . . . . . . . . . . 41 10.2 Computing the inverse Laplace transform . . . . . . . . . . . . . . . . . . . . 41 10.3 Solving an ODE in MAPLE via Laplace transforms . . . . . . . . . . . . . . 42 10.4 The Dirac and Heaviside functions . . . . . . . . . . . . . . . . . . . . . . . 43 10.5 dsolve and Laplace transforms . . . . . . . . . . . . . . . . . . . . . . . . . 43

11 First Order Linear Systems 45 11.1 Linear Algebra in MAPLE . . . . . . . . . . . . . . . . . . . . . . . . . . . . 45 11.2 Phase Portraits of Linear Systems of Differential Equations . . . . . . . . . . 47 11.2.1 Exercise - not for credit . . . . . . . . . . . . . . . . . . . . . . . . . 47 11.3 Systems with Complex Characteristic Roots . . . . . . . . . . . . . . . . . . 49 11.4 Systems with Repeated Characteristic Roots . . . . . . . . . . . . . . . . . . 50 11.5 Assignment on Linear Problems . . . . . . . . . . . . . . . . . . . . . . . . . 51

12 First order non-linear systems 52 12.1 The Van der Pol Equation . . . . . . . . . . . . . . . . . . . . . . . . . . . . 52 12.1.1 Unstable Equilibrium point . . . . . . . . . . . . . . . . . . . . . . . 52 12.1.2 The critical case a = 0 . . . . . . . . . . . . . . . . . . . . . . . . . . 53 12.1.3 Stable equilibrium point . . . . . . . . . . . . . . . . . . . . . . . . . 53 12.2 The Predator-Prey System . . . . . . . . . . . . . . . . . . . . . . . . . . . . 54 12.2.1 Computing the equilibrium points. . . . . . . . . . . . . . . . . . . . 54 12.2.2 Computing stability of the equilibrium points. . . . . . . . . . . . . . 54 12.2.3 Plotting the solution . . . . . . . . . . . . . . . . . . . . . . . . . . . 55 12.3 Exercise on non-linear systems . . . . . . . . . . . . . . . . . . . . . . . . . . 55

13 Fourier Sine Series and the Wave Equation 56 13.1 The Fourier Series of a Continuous Function . . . . . . . . . . . . . . . . . . 56 13.2 The Fourier Series of a Discontinuous Function . . . . . . . . . . . . . . . . . 57 13.3 The Solution of the Wave Equation . . . . . . . . . . . . . . . . . . . . . . . 58

A Solutions to Computer Exercises 59 A.1 Assignment 1 answers . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 59 A.1.1 Logistic Model . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 59 A.1.2 Spruce Budworm Model . . . . . . . . . . . . . . . . . . . . . . . . . 59 A.1.3 Falling Raindrop . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 62 A.2 Assignment 2 answers . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 63 A.2.1 Problem 1: Distinct real roots . . . . . . . . . . . . . . . . . . . . . . 63 A.2.2 Problem 2: Complex roots with damping . . . . . . . . . . . . . . . . 63 A.2.3 Problem 3: Distinct real roots . . . . . . . . . . . . . . . . . . . . . . 65 A.2.4 Problem 4: Beating . . . . . . . . . . . . . . . . . . . . . . . . . . . . 65 A.2.5 Problem 5: Resonance and a MAPLE error . . . . . . . . . . . . . . 66 A.2.6 Problem 6: Long term response. . . . . . . . . . . . . . . . . . . . . . 67 A.3 Assignment 3: Inhomogeneous problems. . . . . . . . . . . . . . . . . . . . . 68 A.3.1 Problem 1a): Undetermined Coefficients. . . . . . . . . . . . . . . . . 68 A.3.2 Problem 1b): Variation of Parameters. . . . . . . . . . . . . . . . . . 69 A.3.3 Problem 1c): Direct use of dsolve. . . . . . . . . . . . . . . . . . . . 70 A.3.4 Problem 2: Resonance. . . . . . . . . . . . . . . . . . . . . . . . . . . 71 A.4 Assignment 4 - Systems of First Order Linear Equations . . . . . . . . . . . 74

B Answers to a few problems from Boyce and DiPrima 79 B.1 Chapter 3 - Second Order Problems . . . . . . . . . . . . . . . . . . . . . . . 79 B.2 Chapter 5 - Series Solution . . . . . . . . . . . . . . . . . . . . . . . . . . . . 82 B.3 Chapter 6 - Laplace Transforms . . . . . . . . . . . . . . . . . . . . . . . . . 83 B.4 Chapter 7 - Systems of First Order Equations . . . . . . . . . . . . . . . . . 85

Chapter 1

Introduction

1.1 General Caveat

This book contains the handouts for M302 - Ordinary differential equations that I developed in 1995 (spring and Fall). The following should be noted:

1. These handouts are MAPLE examples. No “group project” or other impressive piece of curriculum development is here (unfortunately)!

2. The handouts are “rough” and accompanied by considerable discussion in the classroom (which does not appear in these notes). Some were done piece-by-piece during the class, others done in a single sitting during class and still others were take home exercises.

3. This was my first year teaching differential equations with MAPLE, and my first use of MAPLE. So I have probably missed the point in a number of areas, and may not have used the most efficient MAPLE commands in all cases.

4. Not recorded in these notes are various demonstrations I gave to the students during the course of the semester (mainly in MATLAB).

1.2 Sources and Acknowledgments

Much of the material in these notes are “computerizations” of material in the text-book Elementary Differential Equations and Boundary Value Problems by Boyce and DiPrima (fifth edition). Frequently the examples are taken directly from the book and reference is made to examples in the book. This book is the course book for M302.

The other main source of material is the Differential Equations Laboratory Notebook by Borrelli, Coleman and Boyce. Unfortunately this notebook is not oriented to MAPLE. Many of the best examples in these notes are MAPLEizations of work in the notebook.

Chapter 2

Unix Account Setup

Welcome to class MATH302-013!

During this semester, you will be using central UNIX systems to complete class assignments. I recommend you use the X-terminal sites around campus since they operate similarly (but not exactly the same) as the Sun workstations in our classroom. However, you only NEED to use an X-terminal if you want to obtain graphics on the screen.

2.1 Activating your UD computer account

If you have used one of the computers named Brahms, Chopin, or Strauss in the past, then your username and password are what they were before. If you have never used a central UNIX system, you can get your account activated by taking your student ID card (M–F, 8am–5pm) to a Password Changing Station in the Student Services Building or in 002A Smith Hall.

If your userid is still the original 5-digit number, you should select and change it to a username (e.g., marydoe, dprince) by typing the username command. Details on these operations are in the UNIX Handbook, available at the University Bookstore, and in the UNIX Hands-on Worksheet available in the CNS Reference Library (014 Smith).

2.2 Setting up your UD account

This section shows how to set up your account for this class. This needs to be done regardless of whether you are using an X-terminal, dumb terminal or your own PC to log in to the public computers. Make sure that you type all commands exactly as shown.

• Log on to a campus UNIX machine: Exactly how to do this depends on whether you are using a PC, terminal, X-terminal etc.;

• Set up access to commands: You will only need to do this step once, not every

time you log in. Enter the following command exactly as you see it:

~monk/Class/MATH302XXX/student-setup

where XXX is replaced by 013. This has the effect of making sure that you will have proper access to files, commands, etc. for this class.

2.3 Working on MATH302-013class assignments outside the Math. Lab.

1. To begin: type newgrp 2077. The group/project number for your project is 2077 (if

you forget this number, you can see all your projects by typing chdgrp).

2. Do your work. The files related to this class for your instructor’s account are in the

directory $CLASSHOME. For example, if you are told to print your instructor’s file named instructions, you should refer to it by the full name $CLASSHOME/instructions. This only works after you have typed the newgrp 2077 command (i.e. you have done step 1 above).

3. End your work for this class: type exit. Then, you can either log out or work on

something else.

If at any time you need to make sure that 2077 is the current group, type id and the number which will appear in parenthesis following gid will be the current group. If you get any messages asking for a password or other messages, you are not in the project (maybe you just added the class?). See me.

You will automatically see announcements your instructor posts. To review the contents of files in the MATH302-013’s motd (“message of the day”) directory, use the following commands, called aliases.

What you type What the command displays classinfo The contents of the most recently changed message file in the motd directory classinfo filename The contents of the file named filename in the motd directory listclassinfo A directory listing of the files in the motd directory allclassinfo The filename and first 24 lines of every file in the motd directory

For example, typing classinfo exam1-change will display the contents of monk/Class/MATH302013/motd/exam1- change.

2.4 Using Maple from X-terminals

2.4.1 Activating your computer account

If you have used one of the computers named Brahms, Chopin, or Strauss, in the past, then your username and password are what they were before. If you have not, then take your student ID (M-F 8am-5pm) to a Password Changing Station in the Student Services Building or in 002A Smith Hall. There activate your computer account and set a password.

X-terminal sites

It is best to do your Maple work on an X-terminal. Some of the public X-terminal sites are

009B Willard Hall - 23 X-terminals; M-R 8am-10pm, F 8am-5pm, Sat 12noon- 5pm, Sun 12noon-5pm.

114 Newark/Pearson Hall - 24 X-terminals; M-R 9am-midnight, F 9am-5pm, Sat 12noon-4pm, Sun 6pm-midnight.

002 Smith Hall - 28 X-terminals; daily 8am-midnight

2.4.2 Setting up the M302 environment

The following must be done only once this semester. At one of the X-terminal sites • login using your username/usernumber and password.

• move mouse to blue part of the screen and click once the right mouse button. In menu which pops up, left click one of the lines which says Brahms, Chopin, or Strauss.

• in new window which pops up, type carefully

~monk/Class/MATH302013/student-setup

(Don’t forget the ~) and hit Return. The environment is setup.

• move mouse to blue part and right click once. In the menu, left click the line Quit and Logout.

2.4.3 Starting Maple

You should have completed the section Setting up the M302 environment.

• login using your username/usernumber and password.

• move mouse to blue part of screen and right click the mouse button. In menu which pops up, left click one of the lines which says Brahms, Chopin, or Strauss.

• in the new window which pops up, type carefully

newgrp 2077 Hit Return
xmaple & Hit Return

Maple should start up - be patient. If you quit Maple accidentally and want to restart it; just type xmaple & in the same window and Maple will restart.

2.4.4 Logging out from an X-terminal

Be sure to save your Maple work (as described in the Tutorial) if you will need it later. Next, move mouse to blue part and right click once. In the menu, left click the line Quit and Logout. If you don’t do this someone else could wipe out your work.

Chapter 3

Integral Curves and Direction Fields

3.1 Introduction

The purpose of this programing assignment is two fold. First we will set up your account to make sure that you can

1. Access Maple

2. Print graphs and text on the printer in the computer lab.

Second we will do a short computer exercise which will show you how to draw integral curves and direction fields.

3.2 Getting started in the math lab

If the screen on the workstation is blank, move the mouse on the mouse pad to light up the screen. You should see a standard X-windows login screen. If you don’t see the prompt, ask the instructor.

Log on to a lab SUN UNIX machine Login as if you were logging into one of the composers (with your usual user-name and password). Then wait. The window system will start up and the screen will turn blue (this takes about a minute). If you don’t manage to login, ask the instructor.

Run MAPLE Now you are ready to use MAPLE. Just move the mouse until the pointer is in the blue background. Then hold down the left mouse button to obtain the applications menu. With the button held down, move the pointer until the MAPLE entry is high-lighted then let go the button. A window should appear announcing MAPLE’s appearance. Move the pointer into the MAPLE window and you are ready to start. Commands for MAPLE are typed directly into this window.

Log-out from the machine You must do this after you have finished the assignment otherwise strangers may be able to access your account and ruin your work. But don’t logout until you have finished the assignment! To logout, simply move the pointer to the blue background and (using the mouse). Then hold down the right mouse button to activate the root menu. Move the pointer down to the Exit Window System and logout entry and release the mouse button. You will be asked to confirm exiting “mwm”. Confirm this. The screen should change to show the “Solaris” sunburst.

3.3 Drawing integral curves.

We shall start by drawing the integral curves like those shown in Figure 2.1.1 on page 17 in your book (this is so you can check that everything worked right). In your book, the general solution to the ODE y^′ = − 1 2 y + 3 2 is shown to be y = 3 + c exp(−x/2)

To draw integral curves of this equation, start MAPLE. The first step is to tell MAPLE to use the differential equation package and then define the right hand side and differential equation:

with(DEtools);
f:=-y/2+3/2;
ode:=diff(y(x),x)=f;

Next we give Maple some starting points. These are points on the integral curves you want to draw (this is a trial and error problem to get the best choice). For this example, a reasonable choice is (type this into MAPLE)

inits:={[-1,-0.5],[0,0],[0,2],[-1,4],[0,5]};

Finally we can draw the integral curves (notice that you don’t have to solve the differential equation, just tell MAPLE the differential equation):

DEplot(ode,[x,y],x=-1..5,inits);

Now try to draw some integral curves for the problem from class y^′ = −3y Here the general solution is y = c exp(−3x)

Choose limits and initial values to obtain a clear picture of how the solutions behave.

Figure 3.1: A schematic illustrating the construction of one arrow of the direction field.

3.4 Drawing direction fields.

Let the rate function f(x, y) in the first order ODE y^′ = f(x, y) be defined in a rectangle R in the x − y plane given by a ≤ x ≤ b and c ≤ y ≤ d. A solution curve of this ordinary differential equation is a function y(x) whose graph is in R and for which y^′(x) = f(x, y(x)) for a ≤ x ≤ b. From calculus we know that if (x_0, y_0) is a point on the solution curve y = y(x) then y^′(x_0) is the slope of tangent line to y = y(x) at x = x_0. We can represent this graphically by drawing a short line segment centered at (x_0, y_0) with slope y^′(x_0) = f(x_0, y_0). We can repeat this process at a number of grid points throughout R and obtain the direction field of the differential equation. If the grid is fine enough, the resulting direction field may “suggest” solution curves in R with the property that the tangent line to the curve at each point lies along the direction field at that point.^1

To draw the direction field for the differential equation in the previous example

y^′ = − 1 2 y + 3 2 you can use the MAPLE dfieldplot routine:

with(plots):

1This description is taken almost verbatim from Differential Equations Laboratory Notebook by Borrelli, Coleman and Boyce.

with(DEtools):
f:=-y/2+3/2;
ode:=diff(y(x),x)=f;
dfieldplot(ode,[x,y],x=-1..5,y=-2..6);

Better still would be to display the direction fields and the integral curves on one plot:

inits:={[-1,-0.5],[0,0],[0,2],[-1,4],[0,5]};
p1:=DEplot(ode,[x,y],x=-1..5,inits):
p2:=dfieldplot(ode,[x,y],x=-1..5,y=-2..6):
display({p1,p2});

An alternative way to get the direction field and integral curves on one plot is to use the MAPLE DEplot1 command. Note that this command uses f rather than ode as an argument.

DEplot1(f,[x,y],x=-1..5,inits,y=-2..6);

Now consider the ordinary differential equation

y^′ + 2 x y = 4x Draw the direction field for this equation using the range −3 ≤ x ≤ 3 and −1 ≤ y ≤ 5. Plot also the integral curves through [1, 2], [1, 1], and [−1, 1].

Now print the resulting graph by selecting Print/Postscript from the File menu on the Maple graph tool. You will then need to open a command tool window and print the resulting file using lpr.

What do you think happens when x = 0 for the differential equation? It turns out that the general solution to the above ODE is

y(x) = x^2 + c x^2

so the solution through (1, 2) has c = 1 (ie the solution is y(x) = x^2 + 1/x^2) and through (1, 1) is y(x) = x^2.

Compare your picture to Figure 2.2.1 of your book. Congratulations! The first day is over. Don’t forget to logout!

Chapter 4 Solving ODE’s in MAPLE

4.1 Solving ODEs in MAPLE exactly

We shall start by solving the example on page 17 in your book (this is so you can check that everything worked right). In your book, the general solution to the ODE y^′ = − 1 2 y + 3 2 is shown to be y = 3 + c exp(−x/2) To obtain this solution in MAPLE, start up MAPLE. Then load the Differential Equations MAPLE library and the plotting library as follows:

with(DEtools):
with(plots):

Next enter the ODE as before

f:=-y/2+3/2;
ode:=diff(y(x),x)=f;

Before solving the differential equation you can, if you wish, tell MAPLE the initial condition. For example here we shall use the initial condition y(0) = y_0 so we enter

ic:=y(0)=y0;

Now use the MAPLE command dsolve to find the general solution soln:=dsolve({ode,ic},y(x));

Compare the above solution to the one given above. Note that in this case MAPLE gives you an explicit solution y(x) = · · ·.

You can now plot integral curves using this solution by choosing different values of y_0 as follows: first you get the right hand side of the equation, then you plot it

phi:=rhs(soln);
plot({subs(y0=0,phi),subs(y0=-1,phi),subs(y0=1,phi)},x=0..5);

You can plot direction fields and integral curves as before (note the use of a colon (:) after the statement p1:=... which avoids a great deal of useless output:

p1:=plot({subs(y0=0,phi),subs(y0=-1,phi),subs(y0=1,phi)},x=0..5):
p2:=dfieldplot(f,[x,y],x=0..5,y=-1..3):
display({p1,p2});

4.2 Solving ODEs implicitly and drawing integral curves

Solutions to separable equations are often implicit functions of the form g(x, y) = c for some constant c. So the MAPLE dsolve program will often return an implicit solution. For example consider the separable ODE y dy dx + (1 + y^2) sin(x) = 0.

which has the general solution 1 2 ln(1 + y^2) − cos(x) = c.

We can find the general solution in MAPLE as follows

As usual we start by loading DEtools and plots and enter the differential equation into MAPLE

with(DEtools):with(plots):
f:=-(1+y^2)*sin(x);
ode:=y*diff(y(x),x)=f;
soln:=dsolve(ode,y(x));

The solution returned is

soln := y( x )^2 = −1 + e^{( 2 cos( x ) )} C1

This is a simple implicit solution of the differential equation (a little different to the one I gave in class).

To plot integral curves, you need to substitute values for _C1 and use the MAPLE program

implicitplot to plot the solution. For example as follows (this may take a little time, so

we will save the implicitplot output for later - note the colon at the end of the statement

p1:=...)

p1:=implicitplot({subs(_C1=2,soln),subs(_C1=1,soln),subs(_C1=.2,soln),
subs(_C1=4,soln),subs(_C1=8,soln),subs(_C1=16,soln)},x=-1..5,y=-1..5):
display(p1);

The results of implicitplot can be used with dfieldplot to create a direction field and integral curve plot as follows (note the division by y since dfieldplot assumes y^′ = f(x, y)!):

p2:=dfieldplot(f/y,[x,y],x=-1..5,y=-1..5):
display({p1,p2});

4.3 Approximate solutions via dsolve

Last time we saw that MAPLE can use dsolve to solve ordinary differential equations. However in many cases dsolve will not be able to solve a differential equation exactly. So you will have to ask MAPLE to generate an approximate solution for you. In MAPLE this is also done using dsolve but with the type=numeric option. Suppose you want to solve the ordinary differential equation y^′ = −2 y x + 4x

with the initial condition y(0.1) = 4.

The first step is to create a solution function as follows: we define the ordinary differential equation and the initial conditions that use dsolve with the numeric option (note that y(x) is used).

ode:=diff(y(x),x)+2*y(x)/x-4*x=0;
ic:=y(0.1)=4;
soln:=dsolve({ode,ic},y(x),type=numeric);

Now you can use soln to evaluate a numerical approximation to y(x) for different values of x:

soln(1);
soln(7);

If you wish to graph the solution, use odeplot from the plots package:

with(plots):
odeplot(soln,[x,y(x)],0.1..6);

Chapter 5

First Assignment (for credit)

This assignment is to be handed in on Friday 09/22.

5.1 Population models.

Logistic Equation. The logistic model of population says that if p(t) is the population at time t, then p^′ = kp − rp^2 Use the values (chosen entirely at random) of k = 2 and r = 1. Draw the direction field for this equation for 0 ≤ x ≤ 7 and 0 ≤ y ≤ 4. Then superimpose on this direction field the solutions computed by MAPLE using the initial data

p(0) = .1 and p(0) = 4 From the direction field plot, if p(0) > 0, what is

lim_{t→∞} p(t) =?

Spruce Budworm Model. The following is a model for the population of spruce bud-worm (and important factor in the timber industry I believe).

p^′ = p ( k − bp − cp L + p^2 )

There are now negative terms: the logistic term −bp^2 which represents resource depletion, and the term −cp^2/(L + p^2) which represents losses due to predators.

Use the parameter values:

b .2 c 5 k 2 L 2

Draw a direction field plot for the model using 0 ≤ x ≤ 5 and 0 ≤ y ≤ 8. Superimpose the integral curves for p(0) = .5, p(0) = 2, p(0) = 4 and p(0) = 8. From the graph, what is the value of lim_{x→∞} p(x) =?

when p(0) > 0. Does the limit depend on the value of p(0)? Have MAPLE draw a graph of y = p ( k − bp − cp L + p^2 )

against p for 0 ≤ p ≤ 8 to help you understand the behavior of the ODE.

5.2 The Falling Raindrop

This example is from Differential Equations Laboratory Notebook by Borrelli, Coleman and Boyce. Consider a raindrop falling towards the earth near the earth’s surface along the local vertical. Neglecting wind, the raindrop will be acted on by air resistance and gravitational forces. If the velocity of the raindrop is v(t) at time t then v satisfies the differential equation v^′ = −g − k m v|v| v(0) = v_0.

Here g = 9.8 m/s^2 is the force due to gravity on a unit mass. m is the mass of the raindrop in kilograms and k > 0 is the drag coefficient. The initial velocity is v_0 and the term v|v| arises since the force due to drag is proportional to v^2 and always acts to oppose motion (whatever the sign of v).

The first step in analyzing an initial value problem of this type is to “non-dimensionalize” by changing to dimensionless quantities. This has the advantage of removing lots of coefficients from the equation, and allows us to work with quantities that have magnitudes independent of the particular choice of coefficients.

In this example, if we define s to be dimensionless time then it proves convenient to define s by t = ( m kg )_1/_2 s and dimensionless velocity is denoted w and given by v = ( mg k )_1/_2 w.

Change variables in the equation (ie substitute for v and t using w and s) and show that the differential equation becomes dw ds = −1 − w|w| (Show your working in detail on a piece of paper!).

Draw a direction field plot for 0 ≤ s ≤ 5 and −2 ≤ w ≤ 3 and draw integral curves for w_0 = −2, w_0 = −1.5, w_0 = −1, w_0 = .2, w_0 = 0, w_0 = 1, w_0 = 2 and w_0 = 3 (hand in a hard-copy of the plot and the MAPLE session used to generate it).

What is the value of lim_{s→∞} w(s) =?

regardless of w_0? What does this say about

lim_{t→∞} v(t) =?

Chapter 6

Homogeneous second order ordinary differential equations

6.1 Linearity and independence of solutions of homogeneous problems

6.1.1 Superposition principle

In this section we shall use MAPLE to verify that a linear combination of solutions of a homogeneous linear differential equation is a solution of the differential equation. Consider the differential equations x^2y^{′′} − 2xy^′ + 2y = 0

First we shall verify that the functions y_1(x) = x and y_2(x) = x^2 are solutions of this quation. We enter the differential equation into MAPLE and then substitute the two solutions into the equation:

ode:=x^2*diff(y(x),x$2)-2*x*diff(y(x),x)+2*y(x);
y1:=x;
simplify(subs(y(x)=y1,ode));
y2:=x^2;
simplify(subs(y(x)=y2,ode));

Next we take a general linear combination of y_1 and y_2 and verify it satisfies the equation:

y3:=c1*y1+c2*y2;
simplify(subs(y(x)=y3,ode));

As you can see the general linear combination satisfies the ODE.

If the differential equation is non-linear, it is no longer true in general that a linear combination of two solutions satisfies the differential equation. For example consider the differential equation y^{′′} + (y^′)^2 = 0 (this example is from Differentail equations with applications and historical notes by G.F. Simmons). We first verify as before that y_1(x) = 1 and y_2(x) = ln(x) are solutions of the equation:

restart;
ode:=diff(y(x),x$2)+(diff(y(x),x))^2;
y1:=1;
y2:=ln(x);
simplify(subs(y(x)=y1,ode));
simplify(subs(y(x)=y2,ode));

Next we substitute a general linear combination of these solutions into the equation and verify that in general the linear combination is not a solution of the differential equation

y3:=c1*y1+c2*y2;
simplify(subs(y(x)=y3,ode));

So linearity is essential for the superposition principle.

6.1.2 Linear independence

MAPLE can help you to decide if two solutions of a second order problem are linearly independent since MAPLE can compute the Wronskian for you. Suppose you want to verify that y_1(x) = exp(−2x) and y_2(x) = exp(−3x) are linearly independent solutions of the differential equation y^{′′} + 5y^′ + 6y = 0

The first step is to verify that they are solutions of the equation:

restart;
ode:=diff(y(x),x$2)+5*diff(y(x),x)+6*y(x);
y1:=exp(-2*x);
y2:=exp(-3*x);
simplify(subs(y(x)=y1,ode));
simplify(subs(y(x)=y2,ode));

Next we need to evaluate the Wronskian. MAPLE defines the Wronskian as the matrix of functions and derivatives rather than the determinant of the matrix as in the book (and elsewhere). So we need to take the determinant to get the quantity of interest. To access the determinant routine (called det) we need the MAPLE linear algebra routines which we load from the linalg library. Then we compute the Wronskian:

with(linalg):
Wron:=simplify(det(Wronskian([y1,y2],x)));

Since Wron (the Wronskian from the book) is non zero, the two solutions are independent. If the Wronskian were zero, the solutions would be dependent and not form a fundamental set.

6.2 Solving Second Order Equations

6.2.1 Using the characteristic equation

Maple can help you to compute the roots of the characteristic equation for an ordinary differential equation. Consider the equation

4y^{′′} − 8y^′ + 3y = 0

y(0) = 2 y^′(0) = 1/2

This is Example 3 on page 119 of your book (so you can check your answer is correct). The characteristic equation for this differential equation is

4r^2 − 8r + 3 = 0

We can enter this into Maple and solve it as follows:

4*r^2-8*r+3=0;
solve(‘‘,r);

The solution should be r = 1/2 and r = 3/2. Thus the general solution of the ordinary differential equation is y(x) = a exp(x/2) + b exp(3x/2).

We can enter this into Maple and find a and b using the initial conditions. Notice that we use diff to compute y^′ and subs to substitute x = 0 into y and y^′.

y:=a*exp(x/2)+b*exp(3*x/2);
ic1:=subs(x=0,y)=2;
ic2:=subs(x=0,diff(y,x))=1/2;
solve({ic1,ic2},{a,b});

Now we can assign the computed values to a and b, look at the solution and plot the solution

assign(‘‘);
y;
plot(y,x=0..3);

6.2.2 Using dsolve

An easier way is to get the solution is to use dsolve to compute it directly. Suppose we want to compute the solution of the same second order initial value problem as in the previous section:

4y^{′′} − 8y^′ + 3y = 0

y(0) = 2 y^′(0) = 1/2

We start by entering the differential equation (the restart is there to make sure that previous calculations don’t pollute the current one):

restart;
ode:=4*diff(diff(y(x),x),x)-8*diff(y(x),x)+3*y(x)=0;
ic:=y(0)=2,D(y)(0)=1/2;

Note that in entering the differential equation we have use diff(y(x),x) to represent y^′ and

diff(diff(y(x),x),x) to represent y^{′′}. However, in setting the initial conditions we write

y^′(0) as D(y)(0). This inconsistency is just one of Maples delightful quirks. After entering the equation and initial conditions we can compute the solution:

soln:=dsolve({ode,ic},y(x));

The result should be soln := y( x ) = 5 2 e^{( 1/2 x )} − 1 2 e^{( 3/2 x )}

As you can see soln is set to the solution y(x) = · · ·. In order to plot the solution we must extract the right hand side of the equality (i.e. the function). This can by done using the

rhs command on the expression soln). Then we can plot the solution:

y:=rhs(soln);
plot(y,x=0..3);

You can now print a copy of the graph by using the Print command in the File menu. It’s best to choose PostScript as the plotting language. This will create a file called plotoutfile.ps in your file space. You can then print the file by typing the standard UNIX printing command on your system into an a window (not Maple). In the mathematical sciences lab. the command is lpr plotoutfile.ps whereas on the composers it is qpr -q smips plotoutfile.ps (to print the file in Smith).

Note that there is another way of entering the ODE rather than using diff(diff(. You can use

restart;
ode:=4*diff(y(x),x$2)-8*diff(y(x),x)+3*y(x)=0;

The x$2 tells MAPLE that you are using the second derivative.

6.3 Complex roots of the Characteristic equation

The MAPLE routine dsolve can solve second order problems in which the roots of the characteristic equation are complex. Suppose we want to compute the solution of the following second order initial value problem (Example 3 on page 140 of your book):

16y^{′′} − 8y^′ + 145y = 0 y(0) = −2 y^′(0) = 1

Let us first find the general solution. We start by entering the differential equation:

ode:=16*diff(diff(y(x),x),x)-8*diff(y(x),x)+145*y(x)=0;

Now dsolve will give you the solution:

soln:=dsolve(ode,y(x));

The result should be

soln := y( x ) = C1 e^{( 1/4 x )} cos( 3 x ) + C2 e^{( 1/4 x )} sin( 3 x )

As you can see soln is set to the solution written using only real functions and coefficients.

Now you can find the constants _C1 and _C2. We substitute the initial conditions into the right hand side and solve for the contants. Then we assign values to the constants:

eq1:=subs(x=0,rhs(soln))=-2;
eq2:=subs(x=0,diff(rhs(soln),x))=1;
solve({eq1,eq2},{_C1,_C2});
assign(");

Finally lets plot the solution:

plot(rhs(soln),x=0..8);

The plot should look like Figure 3.4.2 on page 141. As you can see the solution oscillates back and forth and also the amplitude (or height) of the oscillations grows.

A quicker way to do the problem would have been for dsolve to find the solution directly (rather than via the general solution). For example:

restart;
ode:=16*diff(y(x),x$2)-8*diff(y(x),x)+145*y(x)=0;
ic:=y(0)=-2,D(y)(0)=1;
soln:=dsolve({ode,ic},y(x));
plot(rhs(soln),x=0..8);

Now have MAPLE solve the following problem and plot the solution (the difference with the previous case is that this sign of the coefficient of y^′ has changed):

16y^{′′} + 8y^′ + 145y = 0 y(0) = −2 y^′(0) = 1

This solution should be oscillatory, but with decreasing amplitude.

Finally solve and plot the solution of the following differential equation (with no term involving y^′):

16y^{′′} + 145y = 0 y(0) = −2 y^′(0) = 1

This solution should oscillate with a constant amplitude.

Chapter 7 Second Assignment (for credit)

7.1 Second order linear ordinary differential equations

For each of the following questions hand in a MAPLE session file (printed using File/Print) which shows the commands you used to obtain the requested output. Clearly mark (in pen if you like) the solution for each problem. Obtain a graph of y(x) against x for the indicated interval of time and indicated equations. Hand in a print of the solution graph (mark clearly which graph belongs to which solution!).

The following examples show some of the behavior found in linear second order ordinary differential equations (with constant coefficients). We will analyze these problems during the next few weeks and understand each result. 1. y^{′′} + 5y^′ + 6y = 0 y(0) = 2 y^′(0) = 3 For 0 ≤ x ≤ 3. 2. y^{′′} + 2y^′ + 5y = 0 y(0) = 1 y^′(0) = 0 For 0 ≤ x ≤ 4. 3. 9y^{′′} − 12y^′ + 4y = 0 y(0) = 2 y^′(0) = −1 For 0 ≤ x ≤ 10.

4. y^{′′} + y = 0.5 cos(0.8x) y(0) = 0 y^′(0) = 0 For 0 ≤ x ≤ 100. 5. y^{′′} + y = 0.5 cos(x) y(0) = 0 y^′(0) = 0 For 0 ≤ x ≤ 40. How do you know that the solution obtained by MAPLE is incorrect? Use the numeric option to dsolve to compute a numerical solution to the problem and compare the graph with the standard dsolve solution. 6. y^{′′} + 0.125y^′ + y = 3 cos(2x) y(0) = 0 y^′(0) = 0 For 0 ≤ x ≤ 50.

Chapter 8 Inhomogeneous Linear Problems

For this handout I show Maple’s response to some of the commands since it’s rather difficult to follow without hardcopy.

8.1 The Method of Undetermined Coefficients

Maple can help you to do the Method of Undertermined Coefficients. First you need to have Maple calculate the fundamental set so you can decide what particular solution to guess. For example suppose that the ODE is given by y^{′′} + y = cos(x)/2 We enter it as usual into Maple:

ode:=diff(y(x),x$2)+y(x)=cos(x)/2;

which produces the following output in Maple:

ode := ( ∂^2 ∂x^2 y( x ) ) + y( x ) = 1 2 cos( x )

We want to find the solution of the homogeneous problem, so we need to the right hand side to be zero (hence we select the left hand side using lhs) and set it equal to zero):

Yc:=dsolve(lhs(ode)=0,y(x));

Maple responds with:

Yc := y( x ) = C1 cos( x ) + C2 sin( x ) Note that the right hand side of the inhomogeneous differential equation is a solution of the homogeneous problem so we must make the following guess for the particular solution:

Yp:=x*(A*cos(x)+B*sin(x));

Maple responds with:

Yp := x ( A cos( x ) + B sin( x ) ) Now we substitute this guess into the differential equation

subs(y(x)=Yp,ode);

to obtain ( ∂^2 ∂x^2 x ( A cos( x ) + B sin( x ) ) ) + x ( A cos( x ) + B sin( x ) ) = 1 2 cos( x )

Now we can simplify the result to force the derivatives to be evaluated

eqn:=simplify(");

and obtain eqn := −2 A sin( x ) + 2 B cos( x ) = 1 2 cos( x ) It turns to be necessary to collect all the terms on the left hand side and then solve for A and B

eq:=lhs(eqn)-rhs(eqn)=0;
solve({coeff(lhs(eq),cos(x))=0,coeff(lhs(eq),sin(x))=0},{A,B});

Maple gives the following solution { B = 1 4 , A = 0 }

Now we set A and B to the values calculated previously

assign(");

So now we have the desired general solution

gsoln:=rhs(Yc)+Yp;

which MAPLE reports as follows gsoln := C1 cos( x ) + C2 sin( x ) + 1 4 x sin( x ) Suppose we also want to satisfy y(0) = 0 and y^′(0) = 0. We need to have Maple compute C1 and C2. First we enforce y(0) = 0 and then y^′(0) = 0 eq1:=simplify(subs(x=0,gsoln))=0;
eq2:=simplify(subs(x=0,diff(gsoln,x)))=0;
solve({eq1,eq2},{_C1,_C2});

which gives the uninspiring solution

{ C1 = 0, C2 = 0 }

Finally we can substitute these values into the general solution

soln:=subs(",gsoln);

to get the solution soln := 1 4 x sin( x )

MAPLE can also get the solution directly using dsolve. Here is how

ic:=y(0)=0,D(y)(0)=0;
soln1:=dsolve({ode,ic},y(x));

On our Sun system under Maple V R3 we get

soln1 := y( x ) = 1 2 sin( x ) arctan ( sin( x ) cos( x ) + 1 )

Lets compare the two solutions by plotting them coloring our correct one in red and MAPLE’s

dsolve solution in blue
p1:=plot(soln,x=0..10,color=red):
p2:=plot(rhs(soln1),x=0..10,color=blue):
with(plots):
display({p1,p2});

Note that MAPLEs solution agrees with our solution until x = π. MAPLE’s soluton is incorrect after that because it makes a mistake about evaluating arctan which is a multivalued function.

8.2 Variation of Parameters

MAPLE can also do variation of parameters. Again we will solve the differential equation

y^{′′} + y = 1 2 cos(x)

First we need to calculate the general solution (as in the previous section). If you have already set Yc to the general solution (in the previous section) skip this step. Otherwise type:

ode:=diff(y(x),x$2)+y(x)=cos(x)/2;
Yc:=dsolve(lhs(ode)=0,y(x));

The first step in variation of parameters is to get the fundamental set from the general solution Yc computed above. This is done as follows:

y1:=rhs(subs(_C1=1,_C2=0,Yc));

so y1 := cos( x ) and

y2:=rhs(subs(_C1=0,_C2=1,Yc));

so y2 := sin( x ) Next we set the function g equal to the inhomogeneous term:

g:=rhs(ode);

and Maple shows g := 1 2 cos( x ) We also need the Wronskian, which MAPLE will compute for you if you first load the linalg package

with(linalg):
Wron:=simplify(det(Wronskian([y1,y2],x)));

and Maple correctly reports Wron := 1 Finally we can use the Variation of Parameters formula

Yp:=simplify(-y1*int(y2*g/Wron,x)+y2*int(y1*g/Wron,x));

and obtain a particular solution: Yp := 1 4 x sin( x ).

This is a perfectly good particular solution and could be used to write down the general solution etc.

8.3 Homogeneous Second Order Problems

In this exercise, we shall investigate the effects of changing the damping parameter b in the ordinary differential equation ay^{′′} + by^′ + cy = 0 (The damping parameter is the coefficient in front of the first derivative term). First enter the general second order homogeneous linear differential equation (given above) into MAPLE. Then we will look at the solution for some special choices of the coefficients.

ode:=a*diff(y(x),x$2)+b*diff(y(x),x)+c*y(x)=0;

8.3.1 No damping

The first example has a = 1, b = 0 and c = 1. Since b = 0 there is no damping and with this choice of parameters b^2 − 4ac < 0 so we expect an pure oscillatory solutiom. To compute this, we start by substituing the values for a, b and c:

ode1:=subs(a=1,b=0,c=1,ode);

Next we will choose some arbitrary non-zero initial conditions, y(0) = 1 and y^′(0) = 0:

ic:=y(0)=1,D(y)(0)=0;

Next we can use dsolve to find the solution.

soln1:=dsolve({ode1,ic},y(x));

There is not much need to plot this solution, but we can do so anyway. Remember to use the rhs function to pick the right hand side as the thing to plot.

plot(rhs(soln1),x=0..10);

Here is the solution for another set of initial conditions

ic:=y(0)=0,D(y)(0)=1;
soln1a:=dsolve({ode1,ic},y(x));

Again we can plot the solution. This ordinary differential equation has solutions that oscillate for ever. There is no damping.

plot(rhs(soln1a),x=0..10);

8.3.2 Negative damping

In this case we choose a = 16, b = −8 and c = 145. Since b < 0, this example has solutions that grow in amplitude due to the fact that the roots of the characteristic equation have positive real part. We can input the differential equation as follows:

ode2:=subs({a=16,b=-8,c=145},ode);

All we need to demonstrate the growing solution is to have some arbitrary non-zero initial data so we choose y(0) = 0 and y^′(0) = 1:

ic:=y(0)=0,D(y)(0)=1;

The solution has grows exponentially as you can tell from the next plot:

soln2:=dsolve({ode2,ic},y(x));
plot(rhs(soln2),x=0..10);

As you can see the solution grows from zero...

8.3.3 Positive Damping

The next example has solutions that die away as x increases due to the roots of the characteristic equation having negative real part. The solution is said to be damped. Under damping:

We start by choosing b = 1:

ode3:=subs({a=1,b=1,c=1},ode);

Again the initial conditions are chosen arbitrarily. Here we take y(0) = 1, y^′(0) = 1:

ic:=y(0)=1,D(y)(0)=1;

As we can see from the output of the next command, the solution is multiplied by a decaying exponential and so it tends to zero for large x:

soln3:=dsolve({ode3,ic},y(x));
plot(rhs(soln3),x=0..10);

The magnitude of b effects the decay of the solution. For this value of b (b = 2) there are no sines or cosines in the solution (no long term oscillation). If we increas b we might

Critical damping:

Increasing b causes a more rapid decrease of the solution to zero unless b is increased too far. For the value of b used in this example (b = 2) there are no sines or cosines in the solution (so no long term oscillation).

ode3:=subs({a=1,b=2,c=1},ode);
soln3:=dsolve({ode3,ic},y(x));
plot(rhs(soln3),x=0..10);

The solution decays more rapidly to zero than when b = 1.

Over damping:

If we make b larger the decay of the solution to zero slows down. Here b = 3.

ode4:=subs({a=1,b=3,c=1},ode);
soln4:=dsolve({ode4,ic},y(x));
plot(rhs(soln4),x=0..10);

8.3.4 Phase Angles

Next I want to illustrate phase angles . Let us consider the following undamped homogeneous differential equation with the given initial conditions u^{′′} + 4u = 0 with u(0) = 1 and u^′(0) = 2 which is entered on the usual way:

ode:=diff(u(x),x$2)+4*u=0;
ic:=u(0)=1,D(u)(0)=2;

As usual we can get a solution using dsolve (a correct one!)

soln:=dsolve({ode,ic},u(x));

which Maple shows to be soln := u( x ) = cos( 2 x ) + sin( 2 x ) Now let us plot the solution and compare it to a pure cosine with the same amplitude. You can see that the solution is just a translation of the simple cosine p1:=plot(rhs(soln),x=0..10,color=red):
p2:=plot(sqrt(2)*cos(2*x),x=0..10,color=blue):
with(plots):
display({p1,p2});

The predicted (from class) phase change is −π/4 . If we plot this solution it is agrees exactly with the true solution above

p3:=plot(sqrt(2)*cos(2*x-Pi/4),x=0..10,color=green):
display({p1,p2,p3});

8.4 Assignment on the solution of inhomogeneous second order problems

Hand in a MAPLE worksheet and appropriate graphs for each question.

8.4.1 Inhomogeneous problems

Consider the following initial value problem for a second order ordinary differential equation: y^{′′} + 25y = sin(11x/2) y(0) = 0 y^′(0) = 0 Use MAPLE to find the solution of this problem in the following three ways.

1. Use the method of undetermined coefficients in MAPLE. 2. Use the method of variation of parameters in MAPLE. 3. Use the MAPLE routine dsolve directly.

Plot graphs of each of these solutions for 0 ≤ x ≤ 30. This is an example of beating.

8.4.2 Resonance

By graphing solutions for different values of ω, find a value of ω that maximizes the amplitude of the long terms solution (after initial transients have decayed) for y^{′′} + y^′ + 25y = 10 sin(ωx) y(0) = 0 y^′(0) = 0 Use the general solution for the problem (obtained in MAPLE) to show that your predicted value is (close to) correct.

Chapter 9 Series Solutions

9.1 Series solutions in dsolve

MAPLE can give you the series solution to an ordinary differential equation by using a special option in the dsolve. Suppose you want to find the series solution to y^{′′} − 2xy^′ − 2y = 0. As usual you enter your differential equation and initial conditions as follows:

ode:=diff(y(x),x$2)-2*x*diff(y(x),x)-2*y(x)=0;

Here we shall use general initial conditions (y(0) = a and y^′(0) = b) so you can see the two linearly independent terms in the solution:

ic:=y(0)=a,D(y)(0)=b;

Now you can use dsolve with the ”type=series” flag set. This will cause MAPLE to find a series solution using the default number of terms (up to order 6):

soln:=rhs(dsolve({ode,ic},y(x),type=series));

This solution is a little difficult to understand. We can regroup the terms to clarify that there are two independent solutions:

z:=a*coeff(convert(soln,polynom),a)+b*coeff(convert(soln,polynom),b);

Maple allows you to change the order of the series solution by specifying a different value of the Order parameter: For example to compute more terms (Order=10 rather than the default of 6) you do the following:

Order:=10;
soln10:=rhs(dsolve({ode,ic},y(x),type=series));
z10:=a*coeff(convert(soln,polynom),a)+b*coeff(convert(soln,polynom),b);

9.2 Taking more and more terms

This short demonstration shows how taking succesively more and more terms in a series solution improves thae accuracy of the solution. First we enter a the simple differential equation y^{′′} + 25y = 0 and the initial conditions (chosen arbitrarily) y(0) = 0 and y^′(0) = 1 and compute it’s exact solution (which we could have done more easily be hand!):

ode:=diff(y(x),x$2)+25*y(x)=0;
ic:=y(0)=0,D(y)(0)=1;
exact:=rhs(dsolve({ode,ic},y(x)));

Maple displays the exact solution: exact := 1 5 sin( 5 x ) Now we plot the exact solution in white saving it for later use:

pe:=plot(exact,x=0..1,color=white):

Next we find the series solution using successively higher orders. Note we convert the series solution to a polynomial because we are going to plot it:

First Order=2:

Order:=2;
s2:=convert(rhs(dsolve({ode,ic},y(x),type=series)),polynom);

which Maple shows to e s2 := x now we plot it and save the picture

p2:=plot(s2,x=0..1,color=red):

Next Order=4:

Order:=4;
s4:=convert(rhs(dsolve({ode,ic},y(x),type=series)),polynom);

giving the solution s4 := x − 25 6 x^3 and we plot this one p4:=plot(s4,x=0..1,color=orange):

Next Order=6:

Order:=6;
s6:=convert(rhs(dsolve({ode,ic},y(x),type=series)),polynom);

Giving the series s6 := x − 25 6 x^3 + 125 24 x^5 and plotting:

p6:=plot(s6,x=0..1,color=yellow):

Finally Order=8:

Order:=8; s8:=convert(rhs(dsolve({ode,ic},y(x),type=series)),polynom);

Which gives even more terms: s8 := x − 25 6 x^3 + 125 24 x^5 −

3125 1008 x^7 and we plot this one

p8:=plot(s8,x=0..1,color=green):

Now we can superimpose the plots of the various approximate solutions. Even Order=8 does not give an acceptable solution as far as x = 1!:

with(plots):
display({pe,p2,p4,p6,p8});

Chapter 10 The Laplace Transform

10.1 Computing the Laplace transform

Note that Maple can compute the Laplace transform of simple functions. For example to compute the Laplace transform of cos(3t) you type in Maple:

laplace(cos(3*t),t,s);

Here t is the variable in cos(3 ∗ t) and s is the transform variable. From now on you will be allowed to use Maple for Laplace and inverse Laplace transforms unless explicitly told otherwise in a question.

10.2 Computing the inverse Laplace transform

Maple can help you with the difficult problem of inverting a Laplace transform (ie computing L^{−1}). As an example, let us consider problem 21 on page 289 in which you are two solve the initial value problem: y^{′′} − 2y^′ + 2y = cos(t) y(0) = 1 y^′(0) = 0 If Y (s) = L(y(t)), you can compute (by hand!) that Y is given by

Y (s) = s s2+1 + s − 2

s^2 − 2s + 2 To find y(t) = L^{−1}(Y (s)) type the following commands in Maple:

readlib(laplace):
Y:=(s/(s^{}2+1)+s-2)/(s^{}2-2*s+2);
invlaplace(Y,s,t);

The command readlib(laplace) tells Maple to load the Laplace transform library. Then Y is set to be the given function and finally invlaplace computes the inverse Laplace transform. The first argument s of invlaplace is the argument of the transform Y (s) and the second argument is the argument for y.

10.3 Solving an ODE in MAPLE via Laplace transforms

You can use Maple to solve the entire problem. For simplicity we shall rewrite the differential equation as y^{′′} − 2y^′ + 2y − cos(t) = 0 We start by entering this differential equation (with right hand side subtracted) into Maple as follows:

ode:=diff(y(t),t$2)-2*diff(y(t),t)+2*y(t)-cos(t);

Next we take the Laplace transform of the equation (and right hand side):

lode:=laplace(ode,t,s);

Note that Maple uses y(0) and D(y)(0) to denote the values of y(0) and y^′(0). We substitute the correct values next:

lode:=subs(y(0)=1,D(y)(0)=0,lode);

Next we solve to obtain the Laplace transform of the solution:

Y:=solve(lode=0,laplace(y(t),t,s));

Finally we can invert the Laplace transform to obtain the solution:

y:=invlaplace(Y,s,t);

Exercise (not for credit): Use Maple to help you solve the initial value problem: y^{′′} + 2y^′ + y = 4 exp(−t) y(0) = 2 y^′(0) = −1 This is question 23 on page 289 of your book. Did you get the correct answer?

10.4 The Dirac and Heaviside functions

Maple even knows about the two new functions introduced in Chapter 6 of your book. The unit step function u_c(t) is given by Heaviside(t-c). Thus you should be able to guess what the following commands will do:

plot(Heaviside(t-2),t=0..4);
laplace(Heaviside(t-2),t,s);

Maple also knows about the Dirac δ function. See what happens when you type

readlib (laplace);
invlaplace(1,s,t);
laplace(Dirac(t-1),t,s);

10.5 dsolve and Laplace transforms

You can force dsolve to use the Laplace transform method. This is useful in two cases, firstly if the right hand side involves the special functions Dirac or Heaviside and secondly if their is resonance (remember dsolve fails if the right hand side of the differential equation is a solution of the homogeneous problem. First we demonstrate using dsolve with the Laplace transform on an ordinary problem:

y^{′′} − 2y^′ + 2y = cos(t) y(0) = 0 y^′(0) = 0

which can be solved as follows (soln and soln1 should be the same).

ode:=diff(y(t),t$2)-2*diff(y(t),t)+2*y(t)-cos(t);
ic:=y(0)=0,D(y)(0)=0;
soln:=rhs(dsolve({ode,ic},y(t),method=laplace));
soln1:=simplify(rhs(dsolve({ode,ic},y(t))));

Next we show solving a resonant problem. The Laplace transform method succeeds as the graphics show:

y^{′′} + y = cos(t) y(0) = 0 y^′(0) = 0

which we solve as follows:

ode:=diff(y(t),t$2)+y(t)-cos(t);
ic:=y(0)=0,D(y)(0)=0;
soln:=rhs(dsolve(ode,y(t),method=laplace));
soln:=subs(ic,soln);
soln1:=rhs(dsolve({ode,ic},y(t)));
with(plots):
p1:=plot(soln,t=0..10,color=red):
p2:=plot(soln1,t=0..10,color=blue):
display({p1,p2});

Clearly the red curve is correct and the blue curve can not be a solution of the differential equation (why?). Note that it seems to be better to impose the initial conditions outside

dsolve when using the Laplace method.

Exercise: Use Maple to help you solve the initial value problem:

y^{′′} − 2y^′ + 2y = δ(t) y(0) = 0 y^′(0) = 0

The solution is called the impulse response for the ordinary differential equation. It should be y(t) = exp(t) sin(t)

Curiously, dsolve gives the wrong answer unless you use method=laplace.

Chapter 11

First Order Linear Systems

11.1 Linear Algebra in MAPLE

Maple can compute eigenvalues and eigenvectors to help with your analysis of linear systems (as long as you have at most four differential equations!!). To use MAPLE’s linear algebra package you should use the following two commands. The first loads the linalg package and the second sets Id to be the identity matrix (appropriate to the context).

with(linalg):
alias(Id=&*());

Matrices are entered row by row. The following command enters a 2 × 2 matrix A = ( 2 1 2 2 )

into MAPLE:

A:=matrix(2,2,[[2,1],[2,2]]);

The eigenvalues can then be computed using the eigenvals command, and the two eigenvalues are saved in the variables l1 and l2.

evs:=eigenvals(A);
l1:=evs[1];
l2:=evs[2];

There is an eigenvects command to find the eigenvalues and eigenvectors of a matrix together. But I have had problems with this command (it gives strange output). So I prefer to find the eigenvectors by finding all solutions of (A − λI)x = 0. This is done as follows. Here sp1 is the set of all solutions - in this case just one vector! We pick out this one vector and call it v1. The same thing is done to get v2.

sp1:=nullspace(l1*Id-A);
v1:=sp1[1];
sp2:=nullspace(l2*Id-A);
v2:=sp2[1];

We can check that these eigenvalues and eigenvectors work:

map(simplify,evalm(multiply(A,v1)-l1*v1));
map(simplify,evalm(multiply(A,v2)-l2*v2));

We can try to diagonalize A by forming a matrix from the eigenvectors (notice the transpose since the eigenvectors becomes rows not columns). Then T^{−1}AT produces the diagonal matrix D as we expect:

T:=transpose(matrix(2,2,[evalm(v1),evalm(v2)]));
multiply(inverse(T),multiply(A,T));

11.2 Phase Portraits of Linear Systems of Differential Equations

The package DEtools contains two routines to help you draw phase portraits. They are

DEplot2 and phaseportrait. To draw the phase portrait of x^′ = Ax, we enter the matrix

into MAPLE. Here A = ( 5 −1 3 1 )

(this is the matrix from class). We can the compute the eigenvalues to find the stability and type of the equilibrium point:

with(linalg):with(DEtools):
A:=matrix(2,2,[[5,-1],[3,1]]);
eigenvals(A);

From the eigenvalues, you can try to make predicitions about the stability and type of the equilibrium point (0, 0)^T . Is the equilibrium point stable in this case? What is the type of the equilibrium point?

To draw a phase portrait of the 2 × 2 linear system, phaseportrait allows you to simply pass the matrix to the function. Then you need to tell phaseportrait the dependent variables (here [x, y]), the integration time (this is chosen be trial and error) here we choose t = −2..2. Then there are a list of initial conditions of the form {[t_0, x_0, y_0], [t_1, x_1, y_1], · · ·} and finally graphing limits to make sure we can see the graph:

phaseportrait(A,[x,y],t=-2..2,{[0,0.5,0],[0,1.5,0],[0,1,1],[0,-1,-1],
[0,1,3],[0,-1,-3],[0,-1,0],[0,0,1]},x=-2..2,y=-2..2);

One of the graphs looks a little rough, so we can force a smoother plot by demanding a smaller step size:

phaseportrait(A,[x,y],t=-2..2,{[0,0.5,0],[0,1.5,0],[0,1,1],[0,-1,-1],
[0,1,3],[0,-1,-3],[0,-1,0],[0,0,1]},x=-2..2,y=-2..2,stepsize=0.1);

The command DEplot2 has a similar syntax, but also plots the direction field:

DEplot2(A,[x,y],t=-2..2,{[0,0.5,0],[0,1.5,0],[0,1,1],[0,-1,-1],[0,1,3],
[0,-1,-3],[0,-1,0],[0,0,1]},x=-2..2,y=-2..2);

11.2.1 Exercise - not for credit

Consider the following ordinary differential equation x^′ = Ax with A = ( −3 √ 2 √ 2 −2 )

• First find the eigenvalues of A and say if the equilibrium point is stable or unstable. • Find the eigenvectors of A and hence write down the general solution. Check your result by comparing your results to Example 2 on page 354 of your book.

• Draw a phase portrait of the solution including the two fundamental solutions found above.

11.3 Systems with Complex Characteristic Roots

Use MAPLE to compute the eigenvalues of the matrix A = ( 2 −5 1 −2 )

Note the eigenvalues are a complex conjugate pair. Now draw the phase plane (with direction field) for the differential equation x^′ = Ax. As you can see the solutions are closed orbits around the equilibrium point.

Here are the commands you could have used to answer this question:

with(linalg):
A:=matrix(2,2,[[2,-5],[1,-2]]);
eigenvals(A);
with(DEtools):
DEplot2(A,[x,y],t=-5..5,{[0,0.5,0],[0,1.5,0],[0,0.1,0]},x=-2..2,y=-2..2);

Now repeat this process with A = ( 1 −5 1 −2 )

This time the equilibrium point is a stable spiral point.

Here are the MAPLE commands you need to use:

A:=matrix(2,2,[[1,-5],[1,-2]]);
eigenvals(A);
with(DEtools):
DEplot2(A,[x,y],t=-5..5,{[0,0.5,0],[0,1.5,0],[0,0.1,0]},x=-2..2,y=-2..2);

11.4 Systems with Repeated Characteristic Roots

Use MAPLE to compute the eigenvalues of the matrix A = ( 2 −3/4 1/3 1 )

Note the eigenvalues are a repeated pair. Now draw the phase plane (with direction field) for the differential equation x^′ = Ax.

Here are the commands you could have used to answer this question:

with(linalg):
A:=matrix(2,2,[[2,-3/4],[1/3,1]]);
eigenvals(A);
eigenvects(A);
with(DEtools):
DEplot2(A,[x,y],t=-5..5,{[0,0.5,0],[0,1.5,0],[0,0,-1],[0,0,1],[0,-1,-1],
[0,3/2,1],[0,-3/2,-1]},x=-2..2,y=-2..2,stepsize=0.1);

11.5 Assignment on Linear Problems

Consider the linear homogeneous ordinary differential equation with constant coefficients given by x^′ = Ax. For each of the matrices A given below do the following:

• Use MAPLE to compute the eigenvalues and eigenvectors of the matrix A. Hence say if the equilibrium point x = 0 is a stable or unstable point. Say also if the point is a node, a saddle or a focus.

• Draw a phase plane diagram for each ordinary differential equation. Display at least 6 solutions that show the general trends of solutions around the origin.

• Hand in MAPLE worksheets, the phase-plane diagrams and a clear answer for each of the following systems: 1. A = ( −3/2 1 −1/4 −1/2 ) . 2. A = ( 2 −5 1 −2 ) . 3. A = ( 3 −2 4 −1 ) . 4. A = ( 3 −2 2 −2 ) . 5. A = ( 1 −2 3 −4 ) .

Chapter 12

First order non-linear systems

12.1 The Van der Pol Equation

In class we predicted that the equilibrium point for the Van der Pol equation:

x^′ = y y^′ = a(1 − x^2)y − x

changes stability as a changes sign. Here we will investigate this graphically. Start MAPLE and enter the Van der Pol equations (notice how to enter a system as a vector of equations):

with(DEtools):
fcns:=[x,y];
odes:=[diff(x(t),t)-y(t)=0,diff(y(t),t)+x(t)-a*(1-x(t)^2)*y(t)];

We predicted that for a ≥ 0 and a < 2 the equilibrium point is unstable and we shall test the case a = 0.25.

12.1.1 Unstable Equilibrium point

Type the following to draw a direction field and two solutions (this will take some time to run - try not to get too impatient!):

DEplot2(subs(a=0.25,odes),fcns,t=0..100,{[0,0.01,0],[0,0,5]},
x=-5..5,y=-5..5,stepsize=0.1);

Notice how the trajectory winds out from the equilibrium point as we expect for an unstable center. Far away from the center, non-linear effects take over and the trajectory approaches a closed curve (call a limit cycle) which is a periodic or cyclic solution of the differential equation. A solution starting outside the limit cycle winds in closer and closer to the limit cycle. Thus if you start anywhere (ie any initial condition) except x = 0, you will wind up

(at large time) very close to the limit cycle. The solution is then oscillatory. To see this graph the actual solution x(t) by typing the following (we tell dsolve to find the solution numerically, in which case we must use odeplot rather than plot to plot the solution):

fcns:={x(t),y(t)};
odes:=diff(x(t),t)-y(t)=0,diff(y(t),t)+x(t)-a*(1-x(t)^2)*y(t);
soln:=dsolve({odes[1],subs(a=0.25,odes[2]),x(0)=0,y(0)=5},fcns,
type=numeric);
with(plots);
odeplot(soln,[t,x(t)],0..50,numpoints=200);

As you can see the solution settles down to a periodic motion.

12.1.2 The critical case a = 0

Now what about a = 0? In this case the origin is a center and so our theory tells us nothing. But for the Van der Pol equation it is easy to see that when a = 0

x^′ = y y^′ = −x This is a linear equation with a center at the origin. See what the direction field and some solutions look like by typing:

fcns:=[x,y];
odes:=[diff(x(t),t)-y(t)=0,diff(y(t),t)+x(t)-a*(1-x(t)^2)*y(t)];
DEplot2(subs(a=0,odes),fcns,t=0..100,{[0,0,1],[0,0,2],[0,0,3]},
x=-5..5,y=-5..5,stepsize=0.1);

12.1.3 Stable equilibrium point

Finally we turn to the case a < 0 and we shall choose a = −.25. In this case we expect a stable spiral equilibrium point. Verify this by typing

DEplot2(subs(a=-0.25,odes),fcns,t=0..100,{[0,1,0],[0,2,0],
[0,2.1,0]},x=-5..5,y=-5..5,stepsize=0.1);

Note also what happens to a trajectory starting outside the (now unstable) limit cycle.

12.2 The Predator-Prey System

In this assignment we will analyze the following predator-prey system: x^′ = x(1 − y/2) y^′ = y(−3 + x)/4 We shall do the following:

1. Calculate the equilibrium points 2. Classify the equilibrium points 3. Draw the phase plane

12.2.1 Computing the equilibrium points.

We enter the differential equations as usual and then compute all the equilibrium points by solving the equations x^′ = y^′ = 0 using the MAPLE solve command.

with(DEtools):
fcns:=[x,y];
odes:=[diff(x(t),t)=x*(1-y/2),diff(y(t),t)=y*(-3+x)/4];
solve({rhs(odes[1])=0,rhs(odes[2])=0},{x,y});

The result is two equilibrium points (x, y) = (0, 0) and (x, y) = (3, 2) These could have been found by hand!!

12.2.2 Computing stability of the equilibrium points.

Now for each equilibrium point, we need to compute the Jacobian (using the MAPLE command jacobian) at the equilibrium point, then compute the eigenvalues to determine the type and stability of the equilibrium point. This is done as follows, first for (x, y) = (0, 0)

with(linalg):
J:=jacobian([rhs(odes[1]),rhs(odes[2])],[x,y]);
J1:=subs(x=0,y=0,evalm(J));
eigenvals(J1);

Thus (x, y) = (0, 0) is a saddle point (we could also compute the eigenvectors to find out the stable and unstable directions). For the second point (x, y) = (3, 2) we proceed as follows:

J2:=subs(x=3,y=2,evalm(J));
eigenvals(J2);

Since the eigenvalues are r_1 = √ 3i/2 and r_2 = − √ 3i/2 the linearized problem possesses a center and we can not conclude anything about the non-linear problem.

12.2.3 Plotting the solution

Drawing the phase-plane and direction fields: As usual we can use DEplot2 to draw the phase plane and direction field. In other cases you will have to experiment with the range of t (here t=-10..10), the initial conditions (here {[0, 2.5, 2], [0, 2, 2], [0, 1, 2], [0, 0.1, 2]}), the graphing window (here x=0..6,y=0..6) and the output step size (here stepsize=0.1) to obtain a good picture.

DEplot2(odes,fcns,t=-10..10,{[0,2.5,2],[0,2,2],[0,1,2],
[0,0.1,2]},x=0..6,y=0..6,stepsize=0.1);

Note that there are periodic solutions but they are not limit cycles.

Plotting the solution against time: You can also plot the solutions against time in the usual way (but it seems easiest to re-enter the ode!):

fcns:={x(t),y(t)};
odes:=diff(x(t),t)=x(t)*(1-y(t)/2),diff(y(t),t)=y(t)*(-3+x(t))/4;
soln:=dsolve({odes[1],odes[2],x(0)=0,y(0)=5},{x(t),y(t)},type=numeric);
with(plots):
odeplot(soln,[[t,x(t)],[t,y(t)]],0..30,numpoints=200);

12.3 Exercise on non-linear systems

The following population model governs two competing species: x^′ = (1 − x − y)x y^′ = (2 − x − 3y)y

• Find all equilibrium points for this system for which x ≥ 0 and y ≥ 0.

• For each equilibrium point use MAPLE to compute the eigenvalues and eigenvectors of the Jacobian matrix. Hence say if the equilibrium point is a stable or unstable point. Say also if the point is a node, a saddle or a focus.

• Draw a phase plane diagram for the ordinary differential equation. Display at least 8 solutions that show the general trends of solutions.

• With this choice of model parameters, can the species co-exist? Will one wipe the other out or will both die out?

• Hand in MAPLE worksheets, the phase-plane diagrams and a clear answer.

Chapter 13 Fourier Sine Series and the Wave Equation

In class we solved the wave equation and showed (if c = 1) that the solution is u(x, t) = ∞∑ n=1 (a_n cos(nπt) + b_n sin(nπt)) sin(nπx) (13.1) where a_n and b_n (n = 1, 2, · · ·) have to be determined. If the initial data is u(x, 0) = g(x) and u_t(x, 0) = h(x) where h(x) and g(x) are known functions we proceed as follows: We plug in t = 0 into (13.1) to get_∞ ∑ n=1 a_n sin(nπx) = g(x) and hence (see your class notes) a_n = 2

∫_1 0 g(x) sin(nπx) dx 13.1 The Fourier Series of a Continuous Function

Suppose that g(x) = { x if 0 ≤ x ≤ 1/2 (1 − x) if 1/2 < x ≤ 1 then a_n = 2 (∫_1 /2 0 x sin(nπx) dx + ∫_1 1/2 (1 − x) sin(nπx) dx ) . We evaluate this and look at the series in MAPLE. First we just evaluate the integrals:

am:=simplify(2*(int(x*sin(m*Pi*x),x=0..1/2)+int((1-x)*sin(m*Pi*x),x=1/2..1)));

Now we can define the partial sum gs by

gs:=Sum(am*sin(m*Pi*x),m=1..M);

Notice we have used the “inert” form of the sum command to delay doing the sum until later when we will supply an M and x. Let us plot some partial sums (using M = 1, 10 and 20) as well as the exact function g(x) which we can express conveniently using the Heaviside function:

with(plots):
p1:=plot(subs(M=1,gs),x=0..1,color=red):
p5:=plot(subs(M=5,gs),x=0..1,color=blue):
p20:=plot(subs(M=20,gs),x=0..1,color=green):
g:= (1-Heaviside(x-1/2))*x+Heaviside(x-1/2)*(1-x);
ge:=plot(g,x=0..1,color=white):
display({p5,p1,p20,ge});

As you can see, when M = 20 the series solution is quite close to the “hat” function (displayed in white).

You might wonder what happens if you use values of x outside [0, 1]. Since the sine functions in the series are periodic of period 2π and odd, the hat repeats as odd function with period 2 as shown in the next computation:

plot(subs(M=20,gs),x=-3..3,color=green);

13.2 The Fourier Series of a Discontinuous Function

Now we turn to the time derivative initial data u_t(x, 0) = h(x). If you differentiate (13.1) with respect to t and plug in t = 0 you obtain: ∞∑ n=1 nπb_n sin(nπx) = h(x) from which we see that nπb_n = 2 ∫_1 0 h(x) sin(nπx) dx In this case we shall suppose that h(x) = { 1 if 0 ≤ x ≤ 3/4 0 if 3/4 < x ≤ 1

then nπb_n = 2 ∫_3/_4 0 sin(nπx) dx. Again we can compute the coefficients b_n as follows bm:=simplify(2*int(sin(m*Pi*x),x=0..3/4)/m/Pi);

and define an approximation to h(x) by

hs:=Sum(m*Pi*bm*sin(m*Pi*x),m=1..M);

Now the convergence of the series to h(x) is much slower as shown in the following plots of partial sums of the Fourier series:

p10:=plot(subs(M=10,hs),x=0..1,color=red):
p20:=plot(subs(M=20,hs),x=0..1,color=blue):
h:=1-Heaviside(x-3/4);
he:=plot(h,x=0..1,color=white):
display({p10,p20,he});

Even adding the first 50 terms doesn’t get you very close to the step function. Notice also the “bumps” close to x = 0 and x = .75. These are always present (for any M but they move closer and closer to the discontinuity as M increases) and are an example of the Gibbs phenomenon:

s50:=evalf(subs(M=50,hs)):
p50:=plot(s50,x=0..1,color=green):
display({p10,p20,p50,he});

13.3 The Solution of the Wave Equation

Now that you have computed the coefficients a_n and b_n you can compute a partial sum approximation to u(x, t) by

u:=Sum((am*cos(m*Pi*t)+bm*sin(m*Pi*t))*sin(m*Pi*x),m=1..M);

With this solution you can plot u at different values of t. For example to plot an approximation to u at t = 0.25 using M = 20 terms you could type:

u25:=evalf(subs(t=.25,M=20,u)):
plot(u25,x=0..1);

You could even make an animation of the solution by typing

ua:=evalf(subs(M=20,u)):
animate(ua,x=0..1,t=0..2,frames=50);

Appendix A Solutions to Computer Exercises

A.1 Assignment 1 answers

A.1.1 Logistic Model

The following are the Maple commands for question 1 (logistic model):

with(DEtools):with(plots):
ode:=diff(p(t),t)=2*p(t)-p(t)^2;

which produces the output: ode := ∂ ∂t p( t ) = 2 p( t ) − p( t )^2 Then

p1:=dfieldplot(ode,[t,p],t=0..7,p=0..4):
p2:=DEplot(ode,[t,p],t=0..7,{[0,.1],[0,4]}):
display({p1,p2});

This creates the graph in Figure A.1. From the graph we can see that if p(0) > 0 the p(t) → 2 as t → ∞. A.1.2 Spruce Budworm Model

Maple commands for the second question (spruce budworm model):

restart;with(plots):with(DEtools):
ode:=diff(p(t),t)=p(t)*(2-.2*p(t)-5*p(t)/(2+p(t)^2));

Figure A.1: Direction field and selected solutions for the logistic equation

which produces the output:

ode := ∂ ∂t p( t ) = p( t ) ( 2 − .2 p( t ) − 5 p( t ) 2 + p( t )^2 )

Then

p1:=dfieldplot(ode,[t,p],t=0..5,p=0..8):
p2:=DEplot(ode,[t,p],t=0..5,{[0,.5],[0,2],[0,4],[0,8]}):
display({p1,p2});

This creates the graph in Figure A.2. To understand the solutions better we perform the following commands:

plot(rhs(ode),p=0..8);
solve(rhs(ode)=0,p(t));

This produces the graph in Figure A.3 and the following solutions of dp/dt = 0:

0, 6.124885419 − .1 10^{−9} I, 1.238442818 + .1 10^{−9} I, 2.636671762 + .1 10^{−9} I

From the two graphs (Figures A.2 and A.3) and above solution, if 0 < p(0) < 2.63667 · · · then p(t) → 1.238 · · · as t → ∞. If p(0) > 2.63667 · · · then p(t) → 6.124885 · · · as t → ∞. If p(0) = 2.63667 · · · then p(t) = 2.63667 · · · for all t (since it is an equilibrium point).

Figure A.2: Direction field and selected solutions for the spruce bud worm equation

Figure A.3: A plot of dp/dt against p for the spruce budworm model.

Figure A.4: Direction field and selected solutions for the falling raindrop model

A.1.3 Falling Raindrop

Maple commands for the last question (falling raindrop):

restart;with(plots):with(DEtools):
ode:=diff(w(s),s)=-1-w(s)*abs(w(s));

which produces the output: ode := ∂ ∂s w( s ) = −1 − w( s ) |w( s )| Then

p1:=dfieldplot(ode,[s,w],s=0..5,w=-2..3):
inits:={[0,-2],[0,-1.5],[0,-1],[0,.2],[0,0],[0,1],[0,2]}:
p2:=DEplot(ode,[s,w],s=0..5,inits):
display({p1,p2});

This creates the graph in Figure A.4. From the graph, for any w(0), we can see that w(s) → −1 as s → ∞. Hence v(t) → √ (mg/k) as t → ∞.

Derivation of the equation

From the fact that t = (m/kg)^{1/2}s we have that dt = (m/kg)^{1/2}ds so the differential equation becomes ( kg m )_1/_2 dv ds = −g − k m v|v| Now using v = (mg/k)^{1/2} we get ( kg m )_1/_2 ( mg k )_1/_2 dw ds = −g − k m (mg k ) w|w| Simplifying this gives the desired differential equation.

A.2 Assignment 2 answers

A.2.1 Problem 1: Distinct real roots

Here are the MAPLE commands for this problem, together with the output and requested graph (see Figure A.5):

with(DEtools):with(plots):
ode1:=diff(y(x),x$2)+5*diff(y(x),x)+6*y(x)=0;

ode1 := ( ∂^2 ∂x^2 y( x ) ) + 5 ( ∂ ∂x y( x ) ) + 6 y( x ) = 0

soln1:=dsolve({ode1,y(0)=2,D(y)(0)=3},y(x));

soln1 := y( x ) = −7 e^{( −3 x )} + 9 e^{( −2 x )}

plot(rhs(soln1),x=0..3);

A.2.2 Problem 2: Complex roots with damping

ode2:=diff(y(x),x$2)+2*diff(y(x),x)+5*y(x)=0;

ode2 := ( ∂^2 ∂x^2 y( x ) ) + 2 ( ∂ ∂x y( x ) ) + 5 y( x ) = 0

soln2:=dsolve({ode2,y(0)=1,D(y)(0)=3},y(x));

soln2 := y( x ) = e^{( −x )} cos( 2 x ) + 2 e^{( −x )} sin( 2 x )

plot(rhs(soln2),x=0..4);

Figure A.5: The graph for assignment 2, problem 1

Figure A.6: The graph for assignment 2, problem 2

Figure A.7: The graph for assignment 2, problem 3

A.2.3 Problem 3: Distinct real roots

ode3:=9*diff(y(x),x$2)-12*diff(y(x),x)+4*y(x)=0;

ode3 := 9 ( ∂^2 ∂x^2 y( x ) ) − 12 ( ∂ ∂x y( x ) ) + 4 y( x ) = 0

soln3:=dsolve({ode3,y(0)=2,D(y)(0)=-1},y(x));

soln3 := y( x ) = 2 e^{( 2/3 x )} − 7 3 e^{( 2/3 x )} x

plot(rhs(soln3),x=0..10);

A.2.4 Problem 4: Beating

ode4:=diff(y(x),x$2)+y(x)=cos(4*x/5)/2;

ode4 := ( ∂^2 ∂x^2 y( x ) ) + y( x ) = 1 2 cos (4 5 x )

Figure A.8: The graph for assignment 2, problem 4

soln4:=dsolve({ode4,y(0)=0,D(y)(0)=0},y(x));

soln4 := y( x ) = 100 9 cos ( 1 5 x )_4 − 100 9 cos ( 1 5 x )_2 + 25 18 − 25 18 cos( x )

plot(rhs(soln4),x=0..100);

A.2.5 Problem 5: Resonance and a MAPLE error

ode5:=diff(y(x),x$2)+y(x)=cos(x)/2;

ode5 := ( ∂^2 ∂x^2 y( x ) ) + y( x ) = 1 2 cos( x )

soln5:=dsolve({ode5,y(0)=0,D(y)(0)=0},y(x));

soln5 := y( x ) = 1 2 sin( x ) arctan ( sin( x ) cos( x ) + 1 )

plot(rhs(soln5),x=0..40);

Figure A.9: The graph for assignment 2, problem 5.

It is not easy to see that the graph in Figure A.9 is incorrect. One cause for concern is that one expects a twice differentiable function for the solution, whereas Figure A.9 has corners. But this might be due to the graphics rather than the presence of real corners. Also, the solution y( x ) = 1 2 sin( x ) arctan ( sin( x ) cos( x ) + 1 )

should not need an arc tangent function (as we know from undetermined coefficients). In practice it is difficult to spot MAPLE’s errors unless one tries a number of different methods of solutions and compares the results to make sure they agree. In this case using a numerical solver gives the “correct” answer:

soln5a:=dsolve({ode5,y(0)=0,D(y)(0)=0},y(x),type=numeric);

soln5a := proc(rkf45_x) ... end

odeplot(soln5a,[x,y(x)],0..40,numpoints=100);

A.2.6 Problem 6: Long term response.

ode6:=diff(y(x),x$2)+diff(y(x),x)/8+y(x)=3*cos(2*x);

Figure A.10: The graph for assignment 2, problem 5 (correct picture!).

ode6 := ( ∂^2 ∂x^2 y( x ) ) + 1 8 ( ∂ ∂x y( x ) ) + y( x ) = 3 cos( 2 x )

soln6:=simplify(expand(dsolve({ode6,y(0)=0,D(y)(0)=0},y(x))));

soln6 := y( x ) = − 16 2465 √ 255 e^{( − 1/16 x )} sin ( 1 16 √ 255 x ) − 288 145 cos( x )^2 + 144 145 + 144 145 e^{( − 1/16 x )} cos ( 1 16 √ 255 x ) + 24 145 sin( x ) cos( x )

plot(rhs(soln6),x=0..50);

A.3 Assignment 3: Inhomogeneous problems.

A.3.1 Problem 1a): Undetermined Coefficients.

First we solve by undetermined coefficients along the lines of the hand-out on this:

with(plots):with(DEtools):
ode1:=diff(y(x),x$2)+25*y(x)=sin(11*x/2);

Figure A.11: The graph for assignment 2, problem 6.

Yc:=dsolve(lhs(ode1)=0,y(x));
Yp:=A*cos(11*x/2)+B*sin(11*x/2);
eq1:=simplify(subs(y(x)=Yp,ode1));
eq1:=lhs(eq1)-rhs(eq1)=0;
solve({coeff(lhs(eq1),cos(11*x/2)),coeff(lhs(eq1),sin(11*x/2))},{A,B});
assign(");
gs1:=rhs(Yc)+Yp;
eq1:=simplify(subs(x=0,gs1))=0;
eq2:=simplify(subs(x=0,diff(gs1,x)))=0;
solve({eq1,eq2},{_C1,_C2});
assign(");
plot(gs1,x=0..30);

A plot of the solution is shown in Figure A.12. This is a typical example of beating.

A.3.2 Problem 1b): Variation of Parameters.

Next we solve the same problem by variation of parameters:

restart;
with(plots):with(DEtools):
ode1:=diff(y(x),x$2)+25*y(x)=sin(11*x/2);

Figure A.12: The graph for assignment 3, problem 1

Yc:=dsolve(lhs(ode1)=0,y(x));
y1:=rhs(subs(_C1=1,_C2=0,Yc));
y2:=rhs(subs(_C1=0,_C2=1,Yc));
g:=rhs(ode1);
with(linalg):
Wron:=simplify(det(Wronskian([y1,y2],x)));
Yp:=simplify(-y1*int(simplify(y2*g/Wron),x)+y2*int(simplify(y1*g/Wron),x),trig);
gs2:=rhs(Yc)+Yp;
eq1:=simplify(subs(x=0,gs2))=0;
eq2:=simplify(subs(x=0,diff(gs2,x)))=0;
solve({eq1,eq2},{_C1,_C2});
assign(");
plot(gs2,x=0..30);

Although the solution looks rather messy, the plot is exactly the same as Figure A.12.

A.3.3 Problem 1c): Direct use of dsolve.

Finally, we solve the same problem by dsolve directly:

restart;
with(plots);with(DEtools);

ode1:=diff(y(x),x$2)+25*y(x)=sin(11*x/2);
soln:=simplify(dsolve({ode1,y(0)=0,D(y)(0)=0},y(x)),trig);
plot(rhs(soln),x=0..30);

Again the solution is messy, but the plot is identical to Figure A.12.

These example illustrate the fact that solutions obtained in MAPLE may not have the simple form you expect. Only by using undetermined cofficients can you be sure that you will get the simple looking solution.

A.3.4 Problem 2: Resonance.

This question asks you to look at the long term solution of a problem. This part of the solution is the particular solution (with solutions of the homogeneous problem removed). The previous question showed that you can get the simplest form of the solution using undetermined coefficients, so we will use that technique here. First we find the general solution:

restart;
with(plots):with(DEtools):
ode:=diff(y(x),x$2)+diff(y(x),x)+25*y(x)=10*sin(omega*x);
Yc:=dsolve(lhs(ode)=0,y(x));
Yp:=A*cos(omega*x)+B*sin(omega*x);
eqn:=simplify(subs(y(x)=Yp,ode));
eqn:=lhs(eqn)-rhs(eqn);
solve({coeff(eqn,cos(omega*x)),coeff(eqn,sin(omega*x))},{A,B});
assign(");
gs:=rhs(Yc)+Yp;
eq1:=simplify(subs(x=0,gs))=0;
eq2:=simplify(subs(x=0,diff(gs,x)))=0;
solve({eq1,eq2},{_C1,_C2});
assign(");
gs;

This gives the following solution to the problem:

y(x) = 10 ω e^{( − 1/2 x )} cos (3 2 √ 11 x ) %1 + 10 33 √ 11 ω (2 ω^2 − 49) e^{( − 1/2 x )} sin ( 3 2 √ 11 x ) %1 − 10 ω cos( ω x ) %1 − 10

(ω^2 − 25) sin( ω x ) %1 %1 := −49 ω^2 + ω^4 + 625

Figure A.13: A graph for assignment 3, problem 2. This shows the amplitude R of the long term response as a function of ω.

The long term solution is

y_p(x) = −10 ω cos( ω x ) −49 ω^2 + ω^4 + 625

− 10

(ω^2 − 25) sin( ω x ) −49 ω^2 + ω^4 + 625 The amplitude R is given by R = √ A^2 + B^2 which we compute next. Then we find where dR/dω = 0 and graph R as a function of ω;

R:=sqrt(A^2+B^2);
solve(diff(R,omega)=0,omega);
plot(R,omega=2..6);

The graph of R as a function of ω is shown in Figure A.13. Clearly, R has a maximum close to ω = 5 and the results of the derivative computation show that the optimal value is ω = √ 49 2 ≈ 4.949747 · · ·

Next we compute the optimal amplitude, and the solution corresponding to this choice of ω. Then we graph the solution:

evalf(subs(omega=sqrt(49/2),R));
gs:=subs(omega=sqrt(49/2),gs);
plot(gs,x=0..20,numpoints=1000);

Figure A.14: A graph for assignment 3, problem 2. This shows the long term response as a function of t for the optimal value of ω.

The plot shown in Figure A.14 confirms that the solution has an optimal amplitude of R_{max} = 20 33 √ 11 ≈ 2.0101 · · ·

Figure A.15: Phase plane and direction field for assignment number four, problem 1

A.4 Assignment 4 - Systems of First Order Linear Equations

1. The MAPLE commands for this question are given next. Here there is no need to compute the eigenvectors for both eigenvalues since the eigenvalues are repeated:

with(DEtools):with(linalg):
alias(Id=&*());
A:=matrix(2,2,[[-3/2,1],[-1/4,-1/2]]);
evs:=eigenvals(A);
sp1:=nullspace(A-evs[1]*Id);
DEplot2(A,[x,y],t=-10..10,{[0,0,5],[0,10,-5],[0,2,1],[0,5,10],[0,-5,-10],
[0,-2,-1]},x=-10..10,y=-10..10,stepsize=0.1);

MAPLE reports that the eigenvalues are λ_1 = λ_2 = −1 and the eigenvector is v^{(1)} = (2, 1)^T . The equilibrium point is an asymptotically stable improper node. The phase plane and direction field (the output from DEplot2) is shown in Figure A.15.

2. The MAPLE commands for this question are given next. Here there is no need to compute the eigenvectors for both eigenvalues since the eigenvectors are complex conjugates of oneanother.

Figure A.16: Phase plane and direction field for assignment number four, problem 2

with(DEtools):with(linalg):
alias(Id=&*());
A:=matrix(2,2,[[2,-5],[1,-2]]);
evs:=eigenvals(A);
sp1:=nullspace(A-evs[1]*Id);
DEplot2(A,[x,y],t=-10..10,{[0,0,1],[0,0,2],[0,0,1/2],[0,0,3]},
x=-10..10,y=-10..10,stepsize=0.1);

MAPLE reports that the eigenvalues are λ_1 = i and λ_2 = −i and the eigenvectors are

v^{(1)} = (2 − i, 1)^T and v^{(2)} = (2 + i, 1)^T . The equilibrium point is an stable center. The

phase plane and direction field (the output from DEplot2) is shown in Figure A.16.

3. The MAPLE commands for this question are given next. Here there is no need to compute the eigenvectors for both eigenvalues since the eigenvectors are complex conjugates of oneanother.

with(DEtools):with(linalg):
alias(Id=&*());
A:=matrix(2,2,[[3,-2],[4,-1]]);
evs:=eigenvals(A);
sp1:=nullspace(A-evs[1]*Id);
DEplot2(A,[x,y],t=-10..10,{[0,0,1],[0,0,2],[0,0,1/2],[0,0,3]},x=-10..10,

Figure A.17: Phase plane and direction field for assignment number four, problem 3

y=-10..10,stepsize=0.1);

MAPLE reports that the eigenvalues are λ_1 = 1+2i and λ_2 = 1−2i and the eigenvectors

are v^{(1)} = (1, 1 − i)^T and v^{(2)} = (1, 1 + i)^T . The equilibrium point is an unstable spiral

point. The phase plane and direction field (the output from DEplot2) is shown in Figure A.17.

4. The MAPLE commands for this question are given next. Here there is we do need to compute the both eigenvectors.

with(DEtools):with(linalg):
alias(Id=&*());
A:=matrix(2,2,[[3,-2],[2,-2]]);
evs:=eigenvals(A);
sp1:=nullspace(A-evs[1]*Id);
sp2:=nullspace(A-evs[2]*Id);
DEplot2(A,[x,y],t=-10..10,{[0,1,2],[0,-1,-2],[0,2,1],[0,-2,-1],[0,-10,-5.5],
[0,-10,-3],[0,10,3],[0,10,5.5]},x=-10..10,y=-10..10,stepsize=0.1);

MAPLE reports that the eigenvalues are λ_1 = 2 and λ_2 = −1 and the eigenvectors are

v^{(1)} = (2, 1)^T and v^{(2)} = (1, 2)^T . The equilibrium point is an unstable saddle point

point. The phase plane and direction field (the output from DEplot2) is shown in Figure A.18.

Figure A.18: Phase plane and direction field for assignment number four, problem 4

5. The MAPLE commands for this question are given next. Here there is we do need to compute the both eigenvectors.

with(DEtools):with(linalg):
alias(Id=&*());
A:=matrix(2,2,[[1,-2],[3,-4]]);
evs:=eigenvals(A);
sp1:=nullspace(A-evs[1]*Id);
sp2:=nullspace(A-evs[2]*Id);
DEplot2(A,[x,y],t=-10..10,{[0,1,1],[0,-1,-1],[0,2,3],[0,-2,-3],[0,-10,-5.5],
[0,10,5.5],[0,0,8],[0,0,-8]},x=-10..10,y=-10..10,stepsize=0.1);

MAPLE reports that the eigenvalues are λ_1 = −1 and λ_2 = −2 and the eigenvectors

are v^{(1)} = (1, 1)^T and v^{(2)} = (−2, 1)^T . The equilibrium point is an asymptotically

stable improper node. The phase plane and direction field (the output from DEplot2) is shown in Figure A.19.

Figure A.19: Phase plane and direction field for assignment number four, problem 5

Appendix B

Answers to a few problems from Boyce and DiPrima

B.1 Chapter 3 - Second Order Problems

Page 142. Number 19. The characteristic equation is r^2 − 2r + 5 = 0 and it’s roots are r_1 = 1 + 2i and r_2 = 1 − 2i. Thus the general soltion is

y(x) = c_1 exp(x) cos(2x) + c_2 exp(x) sin(2x).

To find the constants, we use the initial conditions: y (π 2 ) = c_1 exp ( π 2 ) cos(π) + c_2 exp (π 2 ) sin(π) = −c_1 exp ( π 2 ) = 0.

so c_1 = 0. Using this fact, y^′ ( π 2 ) = c_2 exp ( π 2 ) sin(π) + 2c_2 exp (π 2 ) cos(π) = 2

So c_2 = − exp(−π/2) and y(x) = − exp(x − π/2) sin(2x). The graph looks like (a growing oscillation):

Page 150. Number 19. Use the method of reduction of order. Let y(x) = xv(x), then y^′(x) = v + xv^′ and y^{′′}(x) = 2v^′ + xv^{′′}. Substituting into the differential equation we find xv^{′′} + 4v^′ = 0 So if u(x) = v^′(x) we have xu^′ + 4u = 0 and so u(x) = −3x^{−4} (I have made a particular choice of the constant of integration to get the -3). Then v^′ = −3x^{−4} so v(x) = x^{−3} and hence y_2(x) = xx^{−3} = x^{−2}.

Page 162. Number 14. We seek two particular solutions. The first denoted y_{1p} is for y^{′′}_1 + 4y_1 = x^2 and the second (y_{2p}) is for y^{′′}_2 + 4y_2 = 3 exp(x). Then a particular solution for the full problem is y_p = y_{1p} + y_{2p}.

The characteristic equation is r^2 + 4 = 0 so r_1 = 2i and r_2 = −2i. For y_{1p} we guess

y_{1p}(x) = Ax^2 + Bx + C

And substituting this into the equation for y_1 yields A = 1/4, B = 0 and C = −1/8. For y_{2p} we guess y_{2p}(x) = D exp(x) and find that D = −3/5. Hence

y_p(x) = 1 4 x^2 − 1 8 + 3 5 exp(x)

The general solution is

y(x) = c_1 cos(2x) + c_2 sin(2x) + 1 4 x^2 − 1 8 + 3 5 exp(x).

From the initial condition y(0) = 0 we get c_1 = −3/5 + 1/8 = −19/40 and from y^′(0) = 2 we obtain c_2 = 7/10. Hence the solution is y(x) = − 19 40 cos(2x) + 7 10 sin(2x) + 1 4 x^2 − 1 8 + 3 5 exp(x).

Page 169. Number 14. Substitute y_1(x) = x and y_2(x) = x exp(x) into the differential equation to show they satisfy the differential equation. To get y_p we compute

W [y_1, y_2] = det ( x x exp(x) 1 exp(x) + x exp(x) ) = x^2 exp(x) ≠ 0 if x > 0

hence {x, x exp(x)} is a fundamental set. Using variation of parameters with the above choice of y_1 and y_2 and g(x) = 2x^3/x^2 = 2x gives y_p(x) = xu_1(x) + x exp(x)u_2(x) where

u_1(x) = − ∫ x x exp(x)2x x^2 exp(x) dx = −2x and u_2(x) = ∫ x x2x x^2 exp(x) dx = −2 exp(−x). so y_p(x) = −2x^2 − 2x Since 2x is a solution of the homogeneous problem (2x = 2y_1(x)) we can simplify the particular solution to y_p(x) = −2x^2.

Page 181. Number 9. m = 20g, γ = 400dyne-sec/cm, k = 20 ∗ 980/5 = 3920dyne/cm.

So the equation is 10u^{′′} + 400u^′ + 3920u = 0 or u^{′′} + 20u^′ + 196u = 0. Characteristic equation is r^2 + 20r + 196 = 0 so r_1 = −10 + 4 √ 6i and r_2 = −10 − 4 √ 6i. Hence

u(t) = c_1 exp(−10t) cos(4 √ 6t) + c_2 exp(−10t) sin(4 √ 6t) The initial conditions are u(0) = 2 and u^′(0) = 0 so c_1 = 2 and c_2 = 5 √ 6/6. The solution is

u(t) = 2 exp(−10t) cos(4 √ 6t) + 5 √ 6 6 exp(−10t) sin(4 √ 6t) The quasi-frequency is μ = 4 √ 6 and the quasi-period is T_d = π/(2 √ 6). Hence T_d/T = 7 √ 6/12.

Page 189. Number 11. m = 8/32lb-sec^2/ft. Damping constant γ = 0.25lb-sec/ft. k = 8/(1/2) = 16lb/ft. F (t) = 4 cos(2t)lb. To find the steady state response, we need to find a particular solution of the following ODE u^{′′} + u^′ + 64u = 16 cos(2t) Suppose u_p(t) = A cos(2t) + B sin(2t) then substituting into the equation we have (60A + 2B) cos(2t) + (60B − 2A) sin(2t) = 16 cos(2t) So that A = 240/901 and B = 8/901. Hence u_p(t) = 240 901 cos(2t) + 8 901 sin(2t) This is the long term response of the system since the transient response (solution of the homogeneous problem) dies away due to damping. The amplitude of the long term response is R = F_0 Δ = F_0 √ m^2(ω^2_0 − ω^2)^2 + γ^2ω^2 The maximum amplitude occurs when Δ^2 is a minimum. But Δ^2 = m^2(k/m − ω^2)^2 + γ^2ω^2 so dΔ^2 dm = 0 implies that m = k/ω^2 so m = 16/2^2 = 4 slugs.

B.2 Chapter 5 - Series Solution

Page 227. Number 16. Let y(x) = ∞∑ n=0 a_nx^n so that y^′(x) = ∞∑ n=1 na_nx^{n−1} and y^{′′}(x) = ∞∑ n=2 n(n − 1)a_nx^{n−2}. Substituting into the differential equation: (2 + x^2) ∞∑ n=2 n(n − 1)a_nx^{n−2} − x ∞∑ n=1 na_nx^{n−1} + 4 ∞∑ n=0 a_nx^n = 0 Mutiplying through by the factors outside the sum signs: ∞∑ n=2 2n(n − 1)a_nx^{n−2} + ∞∑ n=2 n(n − 1)a_nx^n − ∞∑ n=1 na_nx^n + ∞∑ n=0 4a_nx^n = 0

Changing index of summation to obtain x^n in each term: ∞∑ n=0 2(n + 2)(n + 1)a_{n+2}x^n + ∞∑ n=2 n(n − 1)a_nx^n − ∞∑ n=1 na_nx^n + ∞∑ n=0 4a_nx^n = 0

Adjusting limits of summation:

4a_2+12a_3x−a_1x+4a_0+4a_1x+ ∞∑ n=2 (2(n + 2)(n + 1)a_{n+2} + n(n − 1)a_n − na_n + 4a_n) x^n = 0

Hence we obtain:

a_2 = a_0, a_3 = −a_1/4, and a_{n+2} = − n^2 − 2n + 4 2(n + 2)(n + 1) a_n, n > 3. Hence the general solution is

y(x) = a_0 ( 1 − x^2 + x^4 6 − x^6 30 + · · · ) + a_1 ( x − x^3 4 + 7x^5 160 + · · · )

Then y(0) = −1 implies a_0 = −1 and y^′(0) = 3 implies that a_1 = 3 so

y(x) = −1 + 3x + x^2 − 3 4 x^3 − 1 6 x^4 + 1 160 x^5 + · · · .

B.3 Chapter 6 - Laplace Transforms

Page 303. Number 10. Let Y (s) = L(y(t)) then

(s^2 + s + 5/4)Y = L(g(t))

But g(t) = sin(t) − u_π(t) sin(t) = sin(t) + u_π(t) sin(t − π) and hence L(g(t)) = 1 s^2 + 1

+ exp(−πs) 1 s^2 + 1 Hence Y (s) = 1 s^2 + s + 5/4 1 s^2 + 1

(1 + exp(−πs))

Then to find L^{−1}Y we use partial fractions: 1 (s^2 + s + 5/4)(s^2 + 1) = As + B s^2 + s + 5/4 + Cs + D s^2 + 1

so that 1 = (As + B)(s^2 + 1) + (Cs + D)(s^2 + s + 5/4) which requires that

A + C = 0, B + D + C = 0, D + A + 5C/4 = 0 and B + 5D/4 = 1 The solution is A = 16/17, B = 12/17, C = −16/17 and D = 4/17 But s^2 + s + 5/4 = (s + 1/2)^2 + 1 so we write 1 (s^2 + s + 5/4)(s^2 + 1) = 16(s + 1/2)/17 + 4/17 (s + 1/2) + 1 + −16s/17 + 4/17 s^2 + 1 Thus L^{−1} ( 1 (s^2 + s + 5/4)(s^2 + 1) ) = 16 17 exp(−t/2) cos(t)+ 4 17 exp(−t/2) sin(t)+− 16 17 cos(t)+ 4 17 sin(t)

If we let h(t) = 16 17 exp(−t/2) cos(t) + 4 17 exp(−t/2) sin(t) + − 16 17 cos(t) + 4 17 sin(t)

then y(t) = h(t) + u_π(t)h(t − π) Page 297. Number 25. Write f using step functions as f(t) = 1 − u_1(t) + u_2(t) − u_3(t) then

L(f(t)) = L(1) − L(u_1(t)) + L(u_2(t)) − L(u_3(t)) = 1 s (1 − exp(−s) + exp(−2s) − exp(−3s)) for s > 0.

Page 303. Number 9. Write g(t) using step functions as g(t) = t − u_1(t)(t − 1) So of Y = L(y(t)) we have (taking Laplace transforms of the differential equation:

(s^2 + 1)Y − 1 = L(t − u_1(t)(t − 1)) =

1 − exp(−s)

s^2 Hence Y (s) = 1 s^2 + 1 ( 1 − exp(−s)

s^2 + 1 ) = 1 s^2 + exp(−s) s^2 + 1 − exp(−s) s^2 From Table 6.2.1 in your book y(t) = L^{−1}(Y (s)) = t − u_1(t)(t − 1 − sin(t − 1))

Page 313. Number 15. Let Y (s) = L(y(t)) then

s^2Y − s + 1 + sY − 1 + 5 4 Y = 1 − exp −πs s So Y = s s^2 + s + 5/4 + 1 − exp(−πs) s(s^2 + s + 5/4) = (s + 1/2) − 1/2 (s + 1/2)^2 + 1 + 1 − exp(−πs) s 1 (s + 1/2)^2 + 1 But L^{−1}(

(s + 1/2) (s + 1/2)^2 + 1

) = exp(−t/2) cos(t)

L^{−1}( −1/2 (s + 1/2)^2 + 1

) = − 1 2 exp(−t/2) sin(t)

L^{−1}( 1 − exp(−πs) s ) = 1 − u_π(t)

L^{−1}( 1 (s + 1/2)^2 + 1

) = exp(−t/2) sin(t)

So by the convolution theorem

y(t) = exp(−t/2) cos(t)− 1 2 exp(−t/2) sin(t)+ ∫ t 0 exp(−(t−τ )/2) sin(t−τ )(1−u_π(τ ) dτ B.4 Chapter 7 - Systems of First Order Equations

Page 344. Number 21. Find the eigenvalues from the characteristic equation:

det(A − λI) = (1 − λ)(λ^2 − 2λ + 5) = 0

Hence λ_1 = 1, λ_2 = 1 + 2i and λ_3 = 1 − 2i. Now we find the eigenvectors using (A − λI)x = 0. First λ_1 = 1 then 0 0 0 2 0 −2 3 2 0 x_1 x_2 x_3 = 0 0 0 Thus x_1 = x_3 and x_1 = −2x_2/3. We can choose x_2 = 3, then x_1 = −2 and x_3 = −2 so

x^{(1)} = −2 3 −2

Next λ_2 = 1 + 2i then

−2i 0 0 2 −2i −2 3 2 −2i x_1 x_2 x_3 = 0 0 0 Thus x_1 = 0 and −ix_2 = x_3. We can choose x_2 = 1, then x_3 = −i so

x^{(2)} = 0 1 −i Finally λ_3 = 1 − 2i then

2i 0 0 2 2i −2 3 2 2i x_1 x_2 x_3 = 0 0 0 Thus x_1 = 0 and ix_2 = x_3. We can choose x_2 = 1, then x_3 = i so

x^{(3)} = 0 1 i Page 349. Number 6. a) The Wronskian W is given by

W [x^{(1)}, x^{(2)}] = det ( t t^2 1 2t ) = 2t^2 − t^2 = t^2

b) Since W vanishes only at t = 0 we know that {x^{(1)}, x^{(2)}} is linearly independent for −∞ < t < 0 or 0 < t < ∞. c) If the differential equation is

x^′ = ( a_{1,1} a_{1,2} a_{2,1} a_{2,2} ) x we know that W^′ = (a_{1,1} + a_{2,2})W so

a_{1,1} + a_{2,2} = 2t/t^2 = 2/t

Hence the coefficient matrix is discontinuous at t = 0. d) Using

x^′ = ( a_{1,1} a_{1,2} a_{2,1} a_{2,2} ) x

we know that ( a_{1,1} a_{1,2} a_{2,1} a_{2,2} ) ( t 1 ) = ( 1 0 ) and ( a_{1,1} a_{1,2} a_{2,1} a_{2,2} ) ( t^2 2t ) = ( 2t 2 )

Thus ( a_{1,1} a_{1,2} a_{2,1} a_{2,2} ) = ( 1 2t 0 2 ) ( t t^2 1 2t )_{−1}

and so ( a_{1,1} a_{1,2} a_{2,1} a_{2,2} ) = ( 0 1 −2/t^2 2/t )

Page 357. Number 1. First find the eigenvalues det ( −2 − λ 1 −5 4 − λ ) = 0 so λ^2 + 2λ − 3 = 0 and λ_1 = 3, λ_2 = −1.

Now find the eigenvectors. First λ = λ_1 = 3 ( −5 1 −5 1λ ) ( ξ_1 ξ_2 ) = ( 0 0 )

so ξ^((1) = (1, 5)^T . Next the eigenvector for λ = λ_2 = −1 ( −1 1 −5 5λ ) ( ξ_1 ξ_2 ) = ( 0 0 )

s ξ^{(2)} = (1, 1, )^T . Thus the general solution is

x(t) = c_1 exp(3t) ( 1 5 ) + c_2 exp(−t) ( 1 1 )

Because x(0) = (1, 3)^T ( 1 3 ) = c_1 ( 1 5 ) + c_2 ( 1 1 )

and c_1 = c_2 = 1/2. The solution is

x(t) = 12 exp(3t) ( 1 5 ) + 1 2 exp(−t) ( 1 1 )

As t → ∞, the solution approaches x_2 = 5x_1 an is unbounded.

Page 369. Number 9. First find the eigenvalues det ( 1 − λ −5 1 −3 − λ ) = 0 so λ^2 + 2λ + 2 = 0 and λ_1 = 3 − 1 + i λ_2 = −1 − i.

Now find the eigenvectors. First λ = λ_1 = −1 + i ( −2 − i −5 1 −2 − iλ ) ( ξ_1 ξ_2 ) = ( 0 0 )

so ξ^((1) = (2+i, 1)^T = (2, 1)^T +i(1, 0^T . There is no need to find the second eigenvector

(it is the conjugate of the one we just found). The general solution is

x(t) = exp(−t) [ c_1 ( cos(t) ( 2 1 ) − sin(t) ( 1 0 )) + c_2 ( sin(t) ( 2 1 ) + cos(t) ( 1 0 ))]

= exp(−t) [ c_1 ( 2 cos(t) − sin(t) cos(t) ) + +c_2 ( 2 sin(t) + cos(t) sin(t) )]

Because x(0) = (1, 1)^T ( 1 1 ) = c_1 ( 2 1 ) + c_2 ( 1 0 )

and c_1 = 1 and c_2 = −1. The solution is

x(t) = exp(−t) ( cos(t) − 3 sin(t) cos(t) − sin(t) )

The solution is an anti-clockwise spiral that approachs the origin as t → ∞.

Page 439. Number 1. Determine the eigenvalues. det ( λ − 3 2 −2 λ + 2 ) = λ^2 − λ − 2 = 0

So λ_1 = −1 and λ_2 = 2.

Now find the eigenvectors. First for λ_1 = −1. ( −4 2 −2 1 ) = 0 so v^{(1)} = ( 1 2 )

Next for λ_2 = 2. ( −1 2 −2 4 ) = 0 so v^{(2)} = ( 2 1 )

So the general solution is

x(t) = c_1 exp(−t) ( 1 2 ) + c_2 exp(2t) ( 2 1 )

The equilibrium point is an unstable saddle point.

Trajectories in the phase plane are shown in the next figure.